Formalization of the Poincaré Disc Model of Hyperbolic Geometry

Danijela Simić, Filip Marić, Pierre Boutry

Noname manuscript No.
(will be inserted by the editor)

Formalization of the Poincaré Disc Model of Hyperbolic Geometry

Danijela Simić · Filip Marić · Pierre Boutry

Received: date / Accepted: date

Abstract We describe formalization of the Poincaré disc model of hyperbolic geometry within the Isabelle/HOL proof assistant. The model is defined within the complex projective line $\mathbb{C}P^1$and is shown to satisfy Tarski's axioms except for Euclid's axiom — it is shown to satisfy it's negation, and, moreover, to satisfy the existence of limiting parallels axiom.

Keywords Poincaré model, Isabelle/HOL, Formalization of geometry

1 Introduction

Poincaré disc is a model of hyperbolic geometry. That fact has been a mathematical folklore for more than 100 years. However, up to the best of our knowledge, fully precise, formal proofs of this fact are lacking. Classic mathematics textbooks on geometry usually show only proof-sketches that lack rigor and often rely on intuition and obviousness from the drawn diagrams. More rigorous exposition of the subject (e.g., [9,40]) usually formalize Klein-Beltrami model and argue that it is much simpler to formalize than the Poincaré disc,

This work is partially supported by the Serbian Ministry of Education and Science grant 174021

Danijela Simić
Faculty of Mathematics, University of Belgrade, Studentski trg 16, 11000 Belgrade, Serbia
E-mail: danijela@matf.bg.ac.rs

Filip Marić
Faculty of Mathematics, University of Belgrade, Studentski trg 16, 11000 Belgrade, Serbia
E-mail: filip@matf.bg.ac.rs

Pierre Boutry
Inria Sophia Antipolis - Méditerranée, 2004, route des Lucioles, BP 93, 06902 Sophia Antipolis Cedex, France
E-mail: pierre.boutry@inria.fr

since h-betweenness relation coincides with the Euclidean betweenness so all axioms that express properties of betweenness are proved easily.

In this paper we present a formalization of the Poincaré disc model in Isabelle/HOL, introduce its basic notions (h-points, h-lines, h-congruence, h-isometries, h-betweenness) and prove that it models Tarski's axioms except for Euclid's axiom. We show that is satisfies the negation of Euclid's axiom, and, moreover, the existence of limiting parallels axiom. The model is defined within the extended complex plane ($\mathbb{C}P^1$), which has been described quite precisely by Schwerdfeger [61] and formalized in the previous work of the first two authors [41]. On the other hand, all proofs about the properties of the model (most notably the proofs that Tarski's axioms are satisfied) are original, and are direct in the sense that they do not rely on any other models. The whole formalization is available from the Archive of Formal Proofs [42, 64].

In this paper we shall present formal definitions of all relevant concepts and formal statements of all relevant lemmas and theorems. We shall also outline their proofs in mathematical prose (often making some simplifications and introducing some minor imprecisions in order to make them more comprehensible). We hope that this exposition of the Poincaré disc could be also interesting even outside the interactive theorem proving community, and that this article could fill in some gaps in the classic literature.

Related work. Although the idea of non-Euclidean geometry (such as hyperbolic geometry that we focus on in this paper) has appeared even in ancient Greece, Saccheri is the first mathematician to have considered the case where Euclid's parallel postulate would not hold [59]. Then, in 1832 and 1840, Bolyai [8] and Lobachevsky [36] published developments about non-Euclidean geometry. Still, more intense study began half a century later. This was most affected by applying the machinery of complex numbers discovered at the end of the 18th century. Complex numbers were an important tool for exploring properties of objects in different geometries. Replacing the Cartesian coordinate plane by complex plane provided simpler formulas to describe geometric objects. After Gaussian theory of curved surfaces [30] and Riemann's work on manifolds [58], the work of Bolyai and Lobachevsky became significant and led to Beltrami's independence proof of the parallel postulate [7]. Hyperbolic geometry is studied through many of its models. The concept of a projective disc model was introduced by Klein while Poincaré investigated the half-plane model proposed by Liouville and Beltrami and primarily studied the isometries of the hyperbolic plane that preserve orientation. In this paper, we focus on the formalization of the latter.

There are many formalizations of fragments of different geometries in the interactive theorem provers. The major part of this research has been devoted to *Euclidean geometry* with probably the largest library [5] being about Tarski's system of geometry [60]. This development was initiated during Narboux's thesis with the formalization of the first eight chapters [46]. Later, the library got extended to contain the results from the first twelve chapters [15], thus allowing Braun and Narboux to mechanize the link from Tarski's axioms to Hilbert's in Coq, Beeson has later written a note [3] to demonstrate that the main results to obtain Hilbert's axioms are contained in [60]. Then, thanks to the formalization of Pappus' theorem [16], the third author, Braun and Narboux completed the formalization of the first part of [60] with the arithmetization of Euclidean geometry [10], thus paving the way for the use of algebraic automated deduction methods in synthetic geometry. Moreover, the library also contains the link from Hilbert's axioms to Tarski's [14] as well as studies about decidability properties [12], parallel postulates [10] and continuity axioms [31].

Other developments based on Tarski's system of geometry have been carried out. For example, Richter, Grabowski and Alama have ported some of these Coq proofs to Mizar (forty-six lemmas) [57]. Moreover, Beeson and Wos proved 200 lemmas of the first twelve chapters of [60] with the Otter theorem prover [6]. Finally, Stojanović-Đurđević, Narboux and Janičić [65] generated automatically some readable proofs in Tarski's system of geometry.

Some formalization of Hilbert's foundations of geometry have been proposed by Dehlinger, Dufourd and Schreck [23] in the Coq proof assistant, and by Dixon, Meikle and Fleuriot [44] using Isabelle/HOL. Dehlinger, Dufourd and Schreck have studied the formalization of Hilbert's foundations of geometry in the intuitionistic setting of Coq [23]. They focus on the first two groups of axioms and prove some betweenness properties. Meikle and Fleuriot have done a similar study within the Isabelle/HOL proof assistant [44]. They went up to twelfth[1] theorem of Hilbert's book. Scott has continued the formalization of Meikle using Isabelle/HOL and revised it [62]. He has corrected some "subtle errors in the formalization of Group III by Meikle". Scott was interested in trying to obtain readable proofs. Later, he developed a system within the HOL-Light proof assistant to automatically fill some gaps in the incidence proofs [63]. Moreover Richter has formalized a substantial number of results based on Hilbert's axioms and a metric axiom system using HOL-Light [56]. Finally, von Plato's constructive geometry [67] has been formalized in Coq by Kahn [35].

Besides Euclidean geometry, *projective geometry* has also been explored using proof assistants. Magaud, Narboux and Schreck proposed alternatives to the traditional axiom systems [20] for plane and space projective geometry based on the notion of ranks and verified using Coq that Desargues' property holds in the latter [39]. The mutual interpretability of their systems with the traditional ones was then formally proved by Braun, Magaud and Schreck in Coq [13].

Despite not being branches of geometry, two fields strongly connected to geometry have been the object of significant formalization efforts: non-standard analysis and computational geometry. *Non-standard analysis* is the field dedicated to the analysis of infinitesimals through hyperreal numbers. Fleuriot formalized notions of non-standard analysis in geometry in Isabelle to mechanize the geometric part of Newton's *Principia* [27] and Kepler's law of Equal

[1] We use the numbering of theorems as of the tenth edition.

Areas [26] using methods of automated theorem proving. Additionally, the discrete model of the continuum known as the Harthong-Reeb line has been formalized in Coq by Magaud, Chollet and Fuchs [38] and in Isabelle by Fleuriot [28]. *Computational geometry* is the study of data structures and algorithms used for solving geometric problems. In Coq, the formalization of combinatorial maps and hypermaps have been carried out by Puitg and Dufourd [55] as well as Dehlinger and Dufourd [22], and Dufourd [24], respectively. These structures have allowed to formally prove the correctness of several algorithms such as the plane Delaunay triangulation algorithm, studied by Dufourd and Bertot [25] in Coq. Furthermore, various convex hull algorithms have also been proved correct by Pichardie and Bertot [54] in Coq, by Meikle and Fleuriot [45] in Isabelle, and by Brun, Dufourd and Magaud [17] in Coq.

There are several formalizations of non-Euclidean geometry. In his MSc thesis, Makarios used Isabelle/HOL to show formally the independence of Euclid's axiom [40], by proving that the Klein–Beltrami model is a model of Tarski's axioms except his Euclidean axiom. The starting point of this work was the formalization of the projective plane $\mathbb{R}P^2$, in which the Klein–Beltrami model is defined. Points are represented by their homogeneous coordinates, as quotients of non-zero vectors in $\mathbb{R}^3$ over the vector proportionality relation. Collineations, invertible linear transformation from the projective plane to itself, are represented by equivalence classes of 3×3 invertible real matrices. The collineations form a group and preserve incidence. Klein–Beltrami model of the hyperbolic plane relies on fixing a bijection between the Cartesian plane and the projective plane with one line removed, and natural bijection (mapping (x, y) to $(x, y, 1)$) is chosen. The set of points of the hyperbolic plane is represented by the open unit disc in $\mathbb{R}^2$, or by its image in the projective plane, according to the natural bijection. Betweenness of the hyperbolic plane is lifted from betweenness in the Cartesian plane. Tarski's betweenness-only axioms, already verified in the Cartesian plane, were easy to lift to equivalent results in the Klein–Beltrami model. However, the definition of congruence depends on collineations that map the unit disc to itself and this definition proved to be difficult to work with. The definition of distance is based on cross-ratio in the projective plane (instead of distance itself, its hyperbolic cosine was used). Although the formalization followed Borsuk and Szmielew's textbook [29], some proofs were missing, some were incomplete and others were difficult to formalize some original proofs had to be invented. Some proofs are rather involved (most notably, the proof that the model satisfies the axiom of segment construction, the five-segments axiom, and the upper 2-dimensional axiom) and they use the definition and properties of perpendicularity (hyperbolic Pythagora's theorem and hyperbolic law of cosines). Since the main goal was to show the independence of Euclid's axiom, and not to investigate hyperbolic geometry, it was not proved that limiting parallels postulate holds in the Klein-Beltrami model. The formalization also does not define circles, line intersection, distance (except its hyperbolic cosine) nor angles.

Based on the work of Makarios, Harrison has shown the independence of Euclid's axiom in HOL-Light, and Coghetto formalized the Klein-Beltrami model within Mizar [18,19].

An exhaustive description of the existing formalizations of geometry is given by Narboux, Janičić and Fleuriot [47].

Outline of the paper. The rest of the paper is structured as follows. In Section 2 we introduce basic concepts of Isabelle/HOL relevant for the present formalization, introduce Tarski's axioms of geometry, and summarize main results of the previous work of the first two authors on formalization of the extended complex plane geometry [41] that are relevant for the present formalization. In Section 3 we define the Poincaré disc model (its points, lines, distance, and betweenness, and prove their properties). In Section 4 we prove that Poincaré model satisfies Tarski's axioms. In Section 5 we draw conclusions and indicate possible directions for further work.

2 Background

2.1 Isabelle/HOL

Isabelle [52,53] is a generic assistant prover that has numerous specializations for different logics, but is most developed for higher-order logic, Isabelle/HOL [50,49]. Formalization of a mathematical theory consists of giving definitions of new concepts (types, constants, functions, etc.) and providing proofs that justify them (lemma, theorems, etc.). Theories are usually described in the language *Isar* [68] which is part of the Isabelle/HOL and which allows writing structured and readable proofs. It is very well adapted for usual mathematical notation. Isabelle/HOL has an extensive library of theories that is constantly increasing and is available through the Archive of Formal Proofs[2].

HOL is a typed logic, with many predefined types available in the library, such as `bool` (Boolean values true and false), `int` (integer numbers), `rat` (rational numbers), `real` (real numbers), and `complex` (complex numbers).

In this paper we shall use real and complex numbers. Imaginary number is denoted by ii. Conversion from real number to complex number is denoted by `cor`, real and imaginary part of a complex number by `Re` and `Im`, the complex conjugate $\overline{z}$ of z by `cnj`, modulus of a complex number $|z|$ by `cmod`, and the complex argument of the number with `arg` (in the Isabelle/HOL is always in the interval $(-\pi, \pi]$). A complex function for the `sgn` determines the complex number on a single circle that has the same argument as well as the given non-zero complex number (i.e, `sgn` $z = z/|z|$). Function `cis` applied to α calculates `cos` α + $ii\cdot$ `sin` α. The determinant of a matrix $|M|$ is denoted by `mat_det`, and the adjoint matrix (conjugate transpose) M^* is denoted by `mat_adj`.

A set of elements of type `'a` is denoted by `'a set`. Difference sets are denoted by $X - Y$, and the image of the function f over the set X is denoted

[2] `http://afp.sourceforge.net/`

with $f`X$. The product type is marked with $\tau_1 \times \tau_2$ (where τ_1 and τ_2 are types of sets).

The type of functions is denoted as $\tau_1 \Rightarrow \tau_2$. Functions are curried and function application within terms is mainly written in prefix form `f x` (instead of $f(x)$, which is closer to the standard mathematical notation), as it is often the case with functional programming.

The terms are λ-expressions with expanded syntax. In spite of the supported type deduction mechanism, it is sometimes necessary explicitly state the type of the term. To express that some term t has τ type, we write $t :: \tau$. The function abstraction be written in form $\lambda x.$ `f` x. Terms also support *let-constructions*: `let` $x = t$ `in` u which is equivalent to u in which all free occurrences are variable x are replaced by t, *if–then–else expressions*: `if` b `then` t_1 `else` t_2, where type for b must be `bool`, and t_1 and t_2 must be the same type, and *case expressions*: `case` e `of` $c_1 \Rightarrow e_1$ | $c_2 \Rightarrow e_2$ | ... | $c_n \Rightarrow e_n$, that has the value e_i if e is equal to some c_i.

Logical formulas are terms of type `bool` and are written in a standard notation. Supported connectives are $\wedge$, $\vee$, $\neg$, $\longrightarrow$, the quantifiers are $\forall$ and $\exists$, and the expression $\exists!\ x.\ P\ x$ denotes that there is exactly one x that meets the predicate P. Also, there are Hilbert's definite and indefinite description operators ($THE\ x.\ P\ x$ denotes the unique value x that satisfies the predicate P).

The `type_synonym` construction introduces a new name for already existing type. *Subtypes* of a given type can be also formed by the `typedef` construction (e.g., the type of $m \times n$ non-degenerate matrices consists of all $m \times n$ matrices that have a non-zero determinant). The *lifting/transfer* package [34] that facilitates defining and using subtypes is available within the main Isabelle/HOL distribution. Functions acting on the subtype are defined by lifting function definitions given on the wider, carrier type (by using commands `setup_lifting`, `lift_definition`). E.g., defining inverse matrix would require giving a definition for every matrix, and then lifting it to non-degenerate matrices where it is well-defined.

Another way to introduce new types, often used in mathematics, are quotient types. For example, positive rational numbers i.e., fractions are a quotient of pairs of natural numbers, where two pairs are identified if they cancel out to the same irreducible fraction. Support for working with quotient types is also provided by the lifting/transfer package. To define a quotient-type one must first define a relation over the carrier type, prove that it is an equivalence relation and then use the `quotient_type` command. Function over quotient types are defined by first defining functions on the carrier type, and then using `lift_definition` command to automatically obtain the definition of the function acting on the quotient type (however, it must be proved that the definition does not depend on the choice of representatives).

Definitions are given using the syntax

definition x `::` τ `where "`x `= ..."`

where x is the constant or the function of the type τ, that is defined. Lemmas are given using syntax:

```
lemma name
  fixes vars
  assumes assms
  shows concl
```

where *name* is the name of the lemma, *vars* conditions that apply, *assms* assumptions, and *concl* is the conclusion of the proposition. If there are no assumptions, the keyword **shows** can be omitted. Also, we will use the syntax **lemma** "$\bigwedge x_1, \ldots x_k.\ [\![asm_1;\ \ldots;\ asm_n]\!] \Longrightarrow concl$" where $asm_1, \ldots, asm_n$ are assumptions, *concl* is a conclusion, and $x_1, \ldots, x_k$ are universally quantified variables. Instead of **lemma**, keyword **theorem** can be used (as there are no formal differences between lemmas and theorems).

Axiom systems in Isabelle/HOL are usually specified as *locales* [2]. A locale is a named context of constants (functions) $f_1, \ldots, f_n$ and assumptions $P_1, \ldots, P_m$ about them that is introduced roughly like

```
locale loc =
  fixes f1, ..., fn
  assumes P1, ..., Pm.
```

Locales can be hierarchical and a locale can extend an existing locale by adding new assumptions, as in **locale** $loc = loc_1 +$ **assumes** .... In the context of a locale, definitions can be made and theorems can be proved. Locales can be interpreted by concrete instances of $f_1, \ldots, f_n$, and then it must be shown that these satisfy assumptions $P_1, \ldots, P_m$. Theorems proved abstractly within a locale are automatically transferred to all their interpretations.

2.2 Tarski's axioms

In 1959 Alfred Tarski gave a set of axioms for a substantial fragment of Euclidean geometry that is formulable in first-order logic with equality, requiring no set-theory. Tarski's axiom system is based on a single primitive type depicting points and two predicates, namely congruence and betweenness. $AB \equiv CD$ states that the segments $\overline{AB}$ and $\overline{CD}$ have the same length. A–B–C means that A, B and C are collinear and B is between A and C (and B may be equal to A or C). For an explanation of the axioms and their history see [66]. Table 1 lists the axioms for Euclidean geometry.

A1	Symmetry	$AB \equiv BA$
A2	Pseudo-Transitivity	$AB \equiv CD \wedge AB \equiv EF \Rightarrow CD \equiv EF$
A3	Cong Identity	$AB \equiv CC \Rightarrow A = B$
A4	Segment construction	$\exists E, A–B–E \wedge BE \equiv CD$
A5	Five-segment	$AB \equiv A'B' \wedge BC \equiv B'C' \wedge$
		$AD \equiv A'D' \wedge BD \equiv B'D' \wedge$
		$A–B–C \wedge A'–B'–C' \wedge A \neq B \Rightarrow CD \equiv C'D'$
A6	Between Identity	$A–B–A \Rightarrow A = B$
A7	Inner Pasch	$A–P–C \wedge B–Q–C \Rightarrow \exists X, P–X–B \wedge Q–X–A$
A8	Lower Dimension	$\exists ABC, \neg A–B–C \wedge \neg B–C–A \wedge \neg C–A–B$
A9	Upper Dimension	$AP \equiv AQ \wedge BP \equiv BQ \wedge CP \equiv CQ \wedge P \neq Q \Rightarrow$
		$A–B–C \vee B–C–A \vee C–A–B$
A10	Euclid	$A–D–T \wedge B–D–C \wedge A \neq D \Rightarrow$
		$\exists XY, A–B–X \wedge A–C–Y \wedge X–T–Y$
A11	Continuity	$\forall \Xi \Upsilon, (\exists A, (\forall XY, \Xi X \wedge \Upsilon Y \Rightarrow A–X–Y)) \Rightarrow$
		$\exists B, (\forall XY, \Xi X \wedge \Upsilon Y \Rightarrow X–B–Y)$

Table 1: Tarski's axiom system for Euclidean geometry.

The symmetry axiom for equidistance (A1 on Tab.1) together with the transitivity axiom for equidistance A2 imply that the equidistance relation is an equivalence relation between pair of points.

The identity axiom for equidistance A3 ensures that only degenerate segments can be congruent to a degenerate segment.

Since we will define the congruence by a formula of the form $f(A, B) = f(C, D)$ for some function f satisfying that $f(x, x) = 0$ for every x, these properties will be easy to verify.



Fig. 1: Axiom of segment construction A4.

The axiom of segment construction A4 allows to extend a segment by a given length (Fig. 1[3]). Here, properties about Möbius transformations will allow to simplify the proof using *without loss of generality* reasoning.

[3] To highlight the fact that all of the axioms except for Euclid's axiom indeed defines a neutral geometry we provide figures both in the Euclidean model and a non-Euclidean model, namely the Poincaré disc model. The figure on the left hand side illustrates the validity of the axiom in Euclidean geometry. The figure on the right hand side either depicts the validity of the statement in the Poincaré disc model or exhibits a counter-example.



Fig. 2: Five-segment axiom A5.

The five-segment axiom A5 corresponds to the well-known Side-Angle-Side postulate but is expressed with the betweenness and congruence relations only. The lengths of $\overline{AB}$, $\overline{AD}$ and $\overline{BD}$ and the fact that A–B–C fix the angle $\angle CBD$ (Fig. 2). For proving that this axiom holds in our model, a crucial point will be that in the Poincaré disc model, the segment congruent to a given segment in the model form a Euclidean circle.

The identity axiom for betweenness A6 expresses that the only possibility to have B between A and A is to have A and B equal. It also insinuates that the relation of betweenness is non-strict, unlike Hilbert's one. As Beeson suggests in [4], this choice was probably made to have a reduced number of axioms by allowing degenerate cases of the Pasch's axiom. This property will follow easily from our definition of betweenness.



Fig. 3: Pasch's axiom A7.

The inner form of Pasch's axiom A7 is the axiom Pasch introduced in [51] to repair the defects of Euclid. It intuitively says that if a line meets one side of a triangle, then it must meet one of the other sides of the triangle. There are three forms of this axiom. Thanks to Gupta's thesis [32], one knows that the inner form and the outer form of this axiom are equivalent and that both of them allow us to prove the weak form. The inner form enunciates Pasch's axiom without any case distinction. Indeed, it indicates that the line BP must meet the triangle ACQ on the side $\overline{AQ}$, as Q is between B and C (Fig. 3). Together with Euclid's axiom, this axiom will be the one requiring the biggest effort as it requires to prove the intersection of h-lines.

The lower two-dimensional axiom A8 asserts that the existence of three non-collinear points. Since our definition of betweenness only allows for h-collinear points to be h-between, this axiom will be proved to hold in our model mostly through calculations.



Fig. 4: Upper-dimensional axiom A9.

The upper two-dimensional axiom A9 means that all the points are coplanar. Since A, B and C are equidistant to P and Q, which are different, they belong to the hyperplane consisting of all the points equidistant to P and Q. Because the upper two-dimensional axiom specifies that A, B and C are collinear, this hyperplane is of dimension one and it fixes the dimension of the space to two. It forbids the existence of the point C' (Fig. 4). As for the five-segment axiom, the fact that the segment congruent to a given segment in the model form a Euclidean circle will prove to be very important.



Fig. 5: Tarski's parallel postulate A10.

Euclid's axiom A10 (Fig. 5) is a modification of an implicit assumption made by Legendre while attempting to prove that Euclid's parallel postulate was a consequence of Euclid's other axioms. According to McFarland, McFarland and Smith [43], the suggestion, made by Gupta [32] and others, that this postulate is due to Lorenz [37] is "doubtful". In fact, the statement to which Gupta refers seems to be the one given in [21] which is indeed different.

While there exist many statements *equivalent* to the parallel postulate, this version is particularly interesting, as it has the advantages of being easily expressed only in term of betweenness, and being valid in spaces of dimension higher than two. However, we will see that a version being expressed with both betweenness and congruence can be easier to work with.

The continuity axiom A11 corresponds to the geometric version of Dedekind's cuts. We should remark that here, Ξ and Υ correspond to first-order formulas. This restriction was made with a view to obtain a very important metatheoretical property for the theory: its completeness. As it could be expected, this restriction weakens the axiom [31]. Again, properties about Möbuis transformations will be helpful to simplify this proof.

Hyperbolic geometry is characterized by the existence of limiting parallels. A ray Aa is a limiting parallel to a ray Bb if they lie on distinct lines not equal to the line AB, they do not meet, and every ray in the interior of the angle BAa meets the ray Bb. In Poincaré model, two rays are limiting parallels iff they meet on the absolute. The existence of limiting parallels axioms states that for each line X_1X_2 and each point A not on that line, there exist two rays, not lying on a same line, that are limiting parallels to rays on the line X_1X_2.



Fig. 6: Existence of limiting parallels

We specify Tarski's axioms in two locales. The first one assumes all axioms except for the Euclid's axiom and the continuity axiom and corresponds to absolute geometry, the second one adds the negation of Euclid's axiom and the existence of limiting parallels to correspond to hyperbolic geometry and the third one then adds the continuity axiom to correspond to elementary hyperbolic geometry.

```
locale TarskiAbsolute =
  fixes cong :: "'p ⇒ 'p ⇒ 'p ⇒ 'p ⇒ bool"
  fixes betw :: "'p ⇒ 'p ⇒ 'p ⇒ bool"
  assumes cong_reflexive: "cong x y y x"
  assumes cong_transitive: "cong x y z u ∧ cong x y v w ⟶ cong z u v w"
```

```
  assumes cong_identity: "cong x y z z ⟶ x = y"
  assumes segment_construction: "∃ z. betw x y z ∧ cong y z a b"
  assumes five_segment:
          "x ≠ y ∧ betw x y z ∧ betw x' y' z' ∧
          cong x y x' y' ∧ cong y z y' z' ∧
          cong x u x' u' ∧ cong y u y' u' ⟶ cong z u z' u'"
  assumes betw_identity: "betw x y x ⟶ x = y"
  assumes Pasch: "betw x u z ∧ betw y v z ⟶ (∃ a. betw u a y ∧ betw x a v)"
  assumes lower_dimension:
          "∃ a. ∃ b. ∃ c. ¬ betw a b c ∧ ¬ betw b c a ∧ ¬ betw c a b"
  assumes upper_dimension:
          "cong x u x v ∧ cong y u y v ∧ cong z u z v ∧ u ≠ v ⟶
          betw x y z ∨ betw y z x ∨ betw z x y"
begin

definition on_line where -- point p is on line ab
  "on_line p a b ⟷ betw p a b ∨ betw a p b ∨ betw a b p"

definition on_ray where -- point p is on ray ab
  "on_ray p a b ⟷ betw a p b ∨ betw a b p"

definition in_angle where -- point p is inside angle abc
  "in_angle p a b c ⟷ b ≠ a ∧ b ≠ c ∧ p ≠ b ∧
                      (∃ x. betw a x c ∧ x ≠ a ∧ x ≠ c ∧ on_ray p b x)"

definition ray_meets_line where -- ray ra-rb meets the line la-lb
  "ray_meets_line ra rb la lb ⟷ (∃ x. on_ray x ra rb ∧ on_line x la lb)"

end

locale TarskiHyperbolic = TarskiAbsolute +
  assumes euclid_negation:
          "∃ a b c d t. betw a d t ∧ betw b d c ∧ a ≠ d ∧
                      (∀ x y. betw a b x ∧ betw a c y ⟶ ¬ betw x t y)"
  assumes limiting_parallels:
          "¬ on_line a x1 x2 ⟹
          (∃ a1 a2. ¬ on_line a a1 a2 ∧
                  ¬ ray_meets_line a a1 x1 x2 ∧ ¬ ray_meets_line a a2 x1 x2 ∧
          (∀ a'. in_angle a' a1 a a2 ⟶ ray_meets_line a a' x1 x2))"

locale ElementaryTarskiHyperbolic = TarskiHyperbolic +
  assumes continuity: "(∃ a. ∀ x. ∀ y. φ x ∧ ψ y ⟶ betw a x y) ⟶
                      (∃ b. ∀ x. ∀ y. φ x ∧ ψ y ⟶ betw x b y)"
```

2.3 Formalization of the extended complex plane

Deep connections between complex numbers and geometry had been well known and carefully studied centuries ago. Fundamental objects that are investigated are the complex plane (usually extended by a single infinite point), its objects (points, lines and circles), and groups of transformations that act on them (e.g., inversions and Möbius transformations).

In the previous work of the first two authors [41], the geometry of the extended complex plane $\overline{\mathbb{C}}$ is formalized as a complex projective line $\mathbb{C}P^1$(by using homogeneous coordinates). It is shown that it is equivalent to the complex plane $\mathbb{C}$ extended by an additional element (treated as the infinite point), and to the Riemann sphere (by means of the stereographic projection [48,61]). In the present formalization we have updated this previous work and simplified many definitions by using lifting/transfer package [34] for both subtypes and quotient types.

2.3.1 Extended complex plane

The *extended complex plane* $\overline{\mathbb{C}}$ is identified with a complex projective line — the one-dimensional projective space over the complex field, sometimes denoted by $\mathbb{C}P^1$. Each point of $\mathbb{C}P^1$is represented by *a pair of complex homogeneous coordinates* (not both equal to zero). Two pairs of homogeneous coordinates represent the same point in $\overline{\mathbb{C}}$ iff they are proportional by a non-zero complex factor. Formalization of this is done in several stages. First, the type synonym for pairs of complex numbers is introduced:

```
type_synonym cvec2 = "complex × complex"
```

Next, equivalence of two pairs of complex numbers is defined (they are equivalent iff they are proportional, where multiplication of the complex vector by a complex scalar is done componentwise).

```
definition eq_cvec2 :: "cvec2 ⇒ cvec2 ⇒ bool" where
  "eq_cvec2 z1 z2 ⟺ (∃ k::complex. k ≠ 0 ∧ z2 = k * z1)"
```

It is rather straightforward to prove that this is an equivalence relation.

Next, the type of non-zero complex vectors is defined as a subtype (these represent homogeneous coordinates of points in $\mathbb{C}P^1$):

```
typedef hc = "{v::cvec2. v ≠ (0, 0)}"
```

The equivalence of pairs of complex numbers is easily lifted to this subtype:

```
setup_lifting type_definition_hc
lift_definition eq_hc :: "hc ⇒ hc ⇒ bool" is eq_cvec2
```

It is easily proved that this is also an equivalence relation.

Finally, the elements of the extended complex plane are defined as equivalence classes of the `eq_hc` relation and the extended complex plane is defined by the following quotient type:

```
quotient_type cp1 = hc / eq_hc
```

The *infinity point* is defined by lifting the homogeneous coordinates $(1, 0)$:

```
definition inf_cvec2 :: "cvec2" ("∞v") where "inf_cvec2 = (1, 0)"
lift_definition inf_hc :: "hc" ("∞hc") is inf_cvec2
lift_definition inf :: "cp1" ("∞h") is inf_hc
```

Each lifting to quotient type requires to prove that the definition does not depend on the choice of representative. These proofs are sometimes tedious, but usually straightforward, so we will not discuss them.

Other distinguished elements of the extended complex plane (0, 1 and i) are defined in the same manner.

The *conversion between the ordinary complex numbers and points in the extended complex plane* is also defined[4].

```
definition of_complex_cvec2 :: "complex ⇒ cvec2"
  where "of_complex_cvec2 z = (z, 1)"
lift_definition of_complex :: "complex ⇒ cp1" is ...
```

Arithmetic operations on ordinary complex numbers (addition, subtraction, multiplication, and division) are extended to the extended complex plane and are defined over homogeneous coordinates. For example, *addition* is defined in the following way.

```
definition add_cvec2 :: "cvec2 ⇒ cvec2 ⇒ cvec2" where
  "add_cvec2 z w = (let (z1, z2) = z; (w1, w2) = w
                    in if z2 ≠ 0 ∨ w2 ≠ 0 then
                          (z1*w2 + w1*z2, z2*w2)
                       else (1, 0))"
lift_definition add :: "cp1 ⇒ cp1 ⇒ cp1" is ...
```

Ratio of complex numbers z, v and w is defined as $\frac{z-v}{z-w}$ and the *cross-ratio* of complex numbers z, u, v, and w as $\frac{(z-u)(v-w)}{(z-w)(v-u)}$. This is extended to homogeneous coordinates and points in the extended complex plane.

```
definition cross_ratio_cvec2 :: "cvec2 ⇒ cvec2 ⇒ cvec2 ⇒ cvec2 ⇒ cvec2" where
  "cross_ratio_cvec2 z u v w =
     (let (z1, z2) = z; (u1, u2) = u; (v1, v2) = v; (w1, w2) = w;
          n1 = z1*u2 - u1*z2; n2 = v1*w2 - w1*v2;
          d1 = z1*w2 - w1*z2; d2 = v1*u2 - u1*v2
      in if n1 * n2 ≠ 0 ∨ d1 * d2 ≠ 0 then
               (n1 * n2, d1 * d2)
          else (1, 1))"
lift_definition cross_ratio :: "cp1 ⇒ cp1 ⇒ cp1 ⇒ cp1 ⇒ cp1" is ...
```

Both operations are very important in the extended complex plane (cross-ratio is a characterizing invariant of Möbius transformations — the fundamental class of transformations of $\overline{\mathbb{C}}$, and it is possible to define lines using ratio since ratio is real iff the three points are on the same line and circles using cross-ratio of points since the cross-ratio is real iff the four points are on the same circle or line). We are going to use cross-ratio in our formalization of the Poincaré disc model and in definition of its central concepts (distance and betweenness).

2.3.2 Möbius transformations

In our formalization *Möbius transformations* are introduced algebraically, as linear transformations of homogeneous coordinates. Each transformation is represented by a regular (non-singular, non-degenerate) 2×2 matrix. Proportional matrices (by a non-zero, complex scalar) represent the same transformation.

[4] From this example it can be seen that the lifting must be done in two stages (one for the subtype and the other for the quotient type). However, to simplify the presentation, in the rest of the paper we shall show only the initial and the final definition.

Again we need to use several stages of definitions. First we introduce the subtype of non-degenerate complex 2×2 matrices (represented by quadruples of complex numbers).

```
typedef cmat2 = "complex × complex × complex × complex"
typedef mmat2 = "{M::cmat2. mat_det M ≠ 0}"
```

Next we say that two complex matrices are equivalent iff they are proportional by a non-zero complex scalar, and then lift this to the subtype of non-degenerate matrices.

```
definition eq_cmat2 :: "cmat2 ⇒ cmat2 ⇒ bool" where
  "eq_cmat2 A B ⟷ (∃ k::complex. k ≠ 0 ∧ B = k * A)"
lift_definition eq_mmat2 :: "mmat2 ⇒ mmat2 ⇒ bool" is ...
```

Finaly, Möbius transformations are introduced by a quotient type over this relation.

```
quotient_type moebius = mmat2 / eq_mmat2
```

Möbius transformations form a group under composition (neutral element corresponds to the identity matrix, inverse element to inverse matrix, and composition corresponds to matrix multiplication). This group is called the *projective general linear group* and denoted by $PGL(2, \mathbb{C})$.

Möbius transformations act on the points of the extended complex plane. A matrix M transforms homogeneous coordinates z to homogeneous coordinates $M \cdot z$. Therefore, we first define actions of matrices to vectors of homogeneous coordinates (by matrix-vector multiplication) and then lift that to quotient types.

```
definition mobius_pt_cmat2_cvec2 :: "cmat2 ⇒ cvec2 ⇒ cvec2" where
  "mobius_pt_cmat2_cvec2 M z = M * z"
lift_definition mobius_pt :: "mobius ⇒ cp1 ⇒ cp1" is ...
```

In the finite complex plane, Möbius transformations are bilinear i.e., transformation defined by the matrix $\begin{pmatrix} a & b \\ c & d \end{pmatrix}$ maps the complex number z to $\frac{az+b}{cz+d}$.

2.3.3 Circlines

The basic object in the extended complex plane is *generalized circle*, or *circline* for short. A significant property of the extended complex plane is that circline represents both line and circles in a uniform way using the equation $Az\overline{z} + B\overline{z} + Cz + D = 0$, where $C = \overline{B}$ and A and D are real. This equation represents a line when $A = 0$ (e.g. when infinite point belongs to the circline) or a circle, otherwise. In our algebraic formalization this equation becomes $z^*Hz = 0$, where z is a vector of homogeneous coordinates, and H is a non-zero, Hermitian matrix $\begin{pmatrix} A & B \\ C & D \end{pmatrix}$ (a matrix is Hermitian if $H^* = H$).

Our formalization again proceeds in three stages. First, the type of Hermitian, non-zero matrices is introduced (as a subtype of all complex 2×2 matrices).

```
definition hermitian :: "cmat2 ⇒ bool" where "hermitian A ⟷ mat_adj A = A"
typedef hmat2 = "{H :: cmat2. hermitian H ∧ H ≠ mat_zero}"
```

Matrices proportional by a real non-zero scalar are considered equivalent.

```
definition real_eq_cmat2 :: "cmat2 ⇒ cmat2 ⇒ bool" where
  "real_eq_cmat2 A B ⟺ (∃ k::real. k ≠ 0 ∧ B = cor k * A)"
lift_definition real_eq_hmat2 :: "hmat2 ⇒ hmat2 ⇒ bool" is ...
```

It is easily shown that this is an equivalence relation, and circlines are defined by a quotient construction as its equivalence classes.

```
quotient_type circline = hmat2 / real_eq_hmat2
```

Among all circlines most prominent ones are the *unit circle*, *x-axis*, and *y-axis*. For example,

```
definition unit_circle_cmat2 :: cmat2 where "unit_circle_cmat2 = (1, 0, 0, −1)"
lift_definition unit_circle :: circline is ...
definition x_axis_cmat2 :: cmat2 where "x_axis_cmat2 = (0, ii, -ii, 0)"
lift_definition x_axis :: circline is ...
```

Each circline defines the *set of points on the circline* (these are the points whose homogeneous coordinates satisfy the relation $z^*Hz = 0$).

```
definition on_circline_cmat2_cvec2 :: "cmat2 ⇒ cvec2 ⇒ bool" where
  "on_circline_cmat2_cvec2 H z ⟷ (vec_cnj z) * H * z = 0"
lift_definition on_circline :: "circline ⇒ cp1 ⇒ bool" is ...
definition circline_set :: "circline ⇒ cp1 set" where
  "circline_set H = { z. on_circline H z }"
```

The sign of the determinant of a representative Hermitian matrix determines the *circline type*. Real circlines have the type -1 and contain infinitely many points, point circlines have the type 0 and contain a single point and imaginary circlines have the type 1 and do not contain any points.

We have also defined oriented circlines. They are also defined by Hermitian matrices, but they satisfy a weaker equivalence relation (they are equivalent iff they are proportional by a positive real scalar).

```
definition pos_eq_cmat2 :: "cmat2 ⇒ cmat2 ⇒ bool" where
  "pos_eq_cmat2 A B ⟺ (∃ k::real. k > 0 ∧ B = cor k * A)"
lift_definition pos_eq_hmat2 :: "hmat2 ⇒ hmat2 ⇒ bool" is pos_eq_cmat2
quotient_type ocircline = hmat2 / pos_eq_hmat2
```

The most important oriented circlines are the *positively oriented unit circle*, *positively oriented x-axis* and *positively oriented y-axis*. For example,

```
lift_definition ounit_circle :: "ocircline" is unit_circle_hmat2
```

Each oriented circline divides the extended complex plane to three sets of points: its interior, its exterior and points on the circline. Interior of an oriented circline is its *disc* and is defined in the following way.

```
definition in_ocircline_cmat2_cvec2 :: "cmat2 ⇒ cvec2 ⇒ bool" where
  "in_ocircline_cmat_cvec H z ⟷ Re ((vec_cnj z) * H * z) < 0"
lift_definition in_ocircline :: "ocircline ⇒ cp1 ⇒ bool" is ...
definition disc :: "circline ⇒ cp1 set" where "disc H = { z. in_ocircline H z }"
```

We shall build the Poincaré model within the unit disc.

```
definition unit_disc :: "cp1 set" where "unit_disc = disc ounit_circle"
```

Angle between two circlines and oriented angle between oriented circlines is defined algebraically, in terms of determinants of circline matrices (for the details we refer the reader to the previous work of the first two authors [41], or Schwerdtfeger [61]).

2.3.4 Möbius action on circlines

Möbius transformations also *act on circlines* and oriented circlines (using matrix *congruence* operation).

```
definition congruence where "congruence M H ≡ mat_adj M * H * M"
definition moebius_cl_cmat2_cmat2 :: "cmat2 ⇒ cmat2 ⇒ cmat2" where
  "moebius_cl_cmat2_cmat2 M H = congruence (mat_inv M) H"
lift_definition moebius_cl :: "moebius ⇒ circline ⇒ circline" is ...
```

The following lemma connects Möbius transformation of points and of circlines (and basically, shows that Möbius transformations map circlines to circlines).

```
lemma "mobius_pt M ‘ circline_set H = circline_set (mobius_cl M H)"
```

Möbius transformations are *conformal* and preserve angle between circlines.

We are specially interested in transformations that map the unit circle to itself. By the fundamental theorem of projective geometry only such transformations that map circlines to circlines are either Möbius transformations or compositions of Möbius transformations and the *conjugation* (its action on the circlines is defined in the following way).

```
fun conjugate_cl_cmat2 :: "cmat2 ⇒ cmat2" where
  "conjugate_cl_cmat2 (A, B, C, D) = (cnj A, cnj B, cnj C, cnj D)"
lift_definition conjugate_cl :: "circline ⇒ circline" is ...
```

Möbius transformations that map the unit circle onto itself can be recognized by their matrices. These matrices are either of the form $k \cdot \begin{pmatrix} a & b \\ \overline{b} & \overline{a} \end{pmatrix}$ or $k \cdot \begin{pmatrix} -a & -b \\ \overline{b} & \overline{a} \end{pmatrix}$, but also have the following characterization.

```
definition unit_circle_fix_cmat2 :: "cmat2 ⇒ bool" where
  "unit_circle_fix_cmat2 M ⟷
     (∃ k::complex. k ≠ 0 ∧ congruence M (1,0,0,−1) = k * (1,0,0,−1))"
lift_definition unit_circle_fix :: "moebius ⇒ bool" is ...

lemma "unit_circle_fix M ⟷ moebius_cl M unit_circle = unit_circle"
```

These transformations form a group. Its important subgroup consists of so-called *Blaschke factors*. The reciprocation can be considered a special case of Blaschke factor (the infinity is mapped to zero, and the unit circle is preserved). Each unit-circle preserving Möbius transformation is a composition of a Blaschke factor (a transformation that brings some point that is not on the unit circle to zero), and a *rotation* around the origin.

```
definition blaschke_cmat2 :: "complex ⇒ cmat2" where
  "blaschke_cmat a = (if cmod a ≠ 1 then (1, -a, -cnj a, 1) else eye)"
lift_definition blaschke :: "complex ⇒ moebius" is ...
definition rotation_cmat2 :: "real ⇒ cmat2" where
  "rotation_cmat2 ϕ = (cis ϕ, 0, 0, 1)"
lift_definition rotation :: "real ⇒ ⇒ moebius" is ...
```

Transformations that preserve the orientation of the unit circle map the unit disc onto itself, while transformations that change the orientation of the unit disc exchange the unit disc and the area outside the unit circle. Central transformations of the Poincaré disc model will be the ones that *map the unit disc onto itself.* Again, they can be recognized from their matrices (the have the form as before, but additionally it must hod that $|a|^2 > |b|^2$).

```
lift_definition unit_disc_fix :: "moebius ⇒ bool" is ...

lemma
assumes "unit_disc_fix M"
shows "moebius_pt M ‘ unit_disc = unit_disc"
```

These transformations also form a group and each of them can be decomposed to a rotation around the origin and a Blaschke factors that maps a point within the unit disc to zero.

```
lemma
assumes "unit_disc_fix M"
shows "∃ k ϕ. cmod a < 1 ∧ M = moebius_rotation ϕ + blaschke a"
```

In the formalization of extended complex plane many other important concepts were formally verified (the chordal metric relevant for elliptic geometry, Euclidean similarities relevant for Euclidean geometry, connections to the Riemann sphere, connections between Möbious transformations and Riemann sphere rotations, circlines connection with circles and lines, relations between Möbius transformations and the orientation) but we are not going to explain those details here and refer the reader to the previous work of the first two authors [41].

2.4 Wlog reasoning

When reason about geometry algebraically in terms of coordinates, often the choice of appropriate coordinate system makes the reasoning much more amenable. We extensively used *without loss of generality* (wlog) technique, described by John Harrison [33].

Wlog reasoning was extensively used in our formalization of the extended complex plane [41]. For example, if a property is preserved under Möbius transformations, then instead of showing that the property holds for any three different points, one can show that the property holds only for points 0_h, 1_h, and ∞_h (looking from another perspective, the coordinate system on the complex projective line can be chosen so that any tree different points have coordinates 0_h, 1_h, and ∞_h). This reasoning is formalized by the following lemma.

```
lemma
  fixes M::moebius
  assumes "P 0h 1h ∞h" and "z1 ≠ z2" and "z2 ≠ z3" and "z1 ≠ z3"
  assumes "⋀ M a b c. P a b c ⟹
                        P (moebius_pt M a) (moebius_pt M b) (moebius_pt M c)"
  shows "P z1 z2 z3"
```

Once this rule applied as a first step in a proof of some statement about three different points, it reduces it to two simpler subgoals (proving the statement for special points 0_h, 1_h, and ∞_h and proving that the property is preserved under Möbius transformations). Automated tactics can be developed to prove preservance under Möbius transformations, but we did not do that in our formalization. Preservance of derived properties is usually proved quite easily by the Isabelle/HOL simplifier, using the set of lemmas that prove preservance for elementary properties (like, for example, incidence).

Wlog reasoning will be one of the central techniques that we will use in the formalization of the Poincaré disc, and we will formulate and use many different wlog lemmas that simplify our proofs (we can freely say that without wlog reasoning our proof would not have been possible). Some of them will be shown in the paper, while others are available in the formal proof documents [64].

3 Formalization of the Poincaré disc

3.1 h-points and h-lines

Basic objects in the Poincare model of the hyperbolic plane are *h-points* and *h-lines*. H-points are points of the extended complex plane that lie within the unit disc.

```
typedef h_point = "{z::cp1. z ∈ unit_disc}"
```

H-lines are circlines that are represented by Hermitian matrices of the form

$$H = \begin{pmatrix} A & B \\ \overline{B} & A \end{pmatrix},$$

for $A \in \mathbb{R}$, and $B \in \mathbb{C}$, and $|B|^2 > A^2$, where the circline equation is the usual one: $z^* H z = 0$. Therefore, each h-line is determined by a real parameter A, and a complex parameter B.

```
definition mk_h_line_cmat2 :: "real ⇒ complex ⇒ cmat2" where
  "mk_h_line_cmat2 A B = (cor A, B, cnj B, cor A)"
```

Such matrix is always Hermitian and is non-zero if one of A or B is non-zero.

We introduce the predicate that checks if a given complex matrix is a matrix of a h-line, and then by means of the lifting package lift it to the type of non-zero Hermitian matrices, and then to circlines (that are equivalence classes of such matrices). Again, lifting requires us to prove that the predicate does not depend on the choice of representative, but that is quite trivial to discharge.

```
definition is_h_line_cmat2 :: "cmat2 ⇒ bool" where
  "is_h_line_cmat2 H ⟷
    (let (A, B, C, D) = H
      in hermitian (A, B, C, D) ∧ A = D ∧ (cmod B)² > (cmod A)²)"
lift_definition is_h_line :: "circline ⇒ bool" is ...
```

We define a subtype of circlines that contains only h-lines, and then lift circline operations to this type. For example, h-incidence is defined by lifting the `on_circline` predicate (that checks if a point in $\mathbb{C}P^1$lies on the given circline).

```
typedef h_line = "{ H. is_h_line H }"
lift_definition h_incident :: "h_line ⇒ h_point ⇒ bool" is on_circline
```

Most lemmas in the further text will be presented in their unlifted form i.e., all our results shall be proved for circlines and points in $\mathbb{C}P^1$and their types (`circline` and `cp1`) and shall contain explicit guards (… ∈ `unit_disc` and `is_h_line` …) in their assumptions and conclusions. By means of the lifting/transfer package their lifting to the `h_point` and `h_line` type becomes almost immediate.

The fundamental characterization of h-lines is that a circline is an h-line iff it is a real (negative-determinant) circline, perpendicular to the unit circle (where perpendicularity of circlines is defined by means of angle between them).

```
lemma shows "is_h_line H ⟷
            circline_type H = -1 ∧ perpendicular H unit_circle"
```

This is an easy consequence of our purely algebraic definition of the angle between circlines and the circline type (both are defined by means of determinants of circline representative matrices). This characterization enables us to switch from algebraic to geometric proofs and use the approach that gives simpler proofs.

The following results are simple consequences of the previous characterization. The x-axis is an h-line, while the unit circle is not. Each h-line contains at least two different points within the unit disc (it contains infinitely many, but showing that it contains two was sufficient for our formalization, and this was surprisingly difficult to show formally).

Every h-line contains the inverse (wrt. the unit circle) of each of its points (note that at most one of them belongs to the unit disc), and is invariant under unit circle inversion. Since there is a unique circline containing every three different points, there is no more than one h-line that contains two different h-points (existence is shown later, constructively). We formally show this only for points within the unit disc, but this could be relaxed (to any two different points except the pair of points zero and infinity). If a h-line contains zero, than it also contains infinity (the inverse point of zero) and is by definition the Euclidean line. Otherwise it must be an Euclidean circle.

H-isometries are conjugation, Möbius transformations that map the unit disc onto itself, and their compositions. For all concepts that we introduce in the model, we show that they are preserved by h-isometries. We show that h-lines are preserved by h-isometries (i.e., that h-isometries map h-lines to h-lines). First we show that all unit circle preserving Möbius transforms map h-lines to h-lines.

```
lemma
assumes "unit_circle_fix M"
shows "is_h_line (moebius_cl M H) ⟷ is_h_line H"
```

This is shown as an easy consequence of the characterization of h-lines as real circlines perpendicular to the unit circle, the fact that Möbius transformations preserve the circline type and are conformal (preserve angles and, therefore, perpendicularity), and that unit disc preserving transforms map unit circle onto itself.

Next we show that conjugation also preserves h-lines.

```
lemma "is_h_line (conjugate_cl H) ⟷ is_h_line H"
```

Unlike for the Möbius transforms, where we used a geometric proof, this is easily shown analytically, by direct calculations.

A set of points is *h-collinear* if there exists an h-line containing all of them.

```
definition h_collinear :: "cp1 set ⇒ bool" where
  "h_collinear S ⟷ (∃ p. is_h_line p ∧ S ⊆ circline_set p)"
```

H-collinearity is preserved by h-isometries (since h-lines are preserved).

Construction of an h-line through the two given points. To show the existence of an h-line through any two given h-points we have formalized the analytic construction. For any two different points u and v in the unit disc, the h-line is given by an h-line matrix with $A = i \cdot (u\overline{v} - v\overline{u})$, and $B = i \cdot (v(|u|^2 + 1) - u(|v|^2 + 1))$. If $A \neq 0$ (meaning that 0, u and v are not collinear), it is an Euclidean circle with center in $-\frac{B}{A}$, i.e. in

$$\frac{u(|v|^2 + 1) - v(|u|^2 + 1)}{u\overline{v} - v\overline{u}}.$$

If $A = 0$ (meaning that 0, u and v are collinear i.e., that there exists a real number k such that $v = k \cdot u$), it is an Euclidean line with normal vector

B. If $u = v$, than $A = B = 0$, which is not a valid circline. Note that this construction can be used also if u and v are points outside the unit disc. The construction fails when u and v are mutually inverse as this would also yield $A = B = 0$ (interestingly, the standard ruler and compass construction also fails in that case). If $\{u, v\} = \{0, \infty\}$, then there is no unique h-line through the two points, although they are different, but in other cases of mutually inverse points the previous construction can be applied on 0 and u yielding valid circline.

We formalized the previous construction in homogeneous coordinates, simplifying cases when $A = 0$, and then lift it to extended complex plane and circlines (by showing that the construction always yields a non-zero, Hermitian matrix, and then by showing that it does not depend on the choice of representative vectors of homogeneous coordinates).

```
definition h_line_cvec2_cmat2 :: "cvec2 ⇒ cvec2 ⇒ cmat2" where
  "h_line_cvec2_cmat2 u v =
    (let (u₁, u₂) = u; (v₁, v₂) = v;
         nom = v₁*cnj v₂*(u₁*cnj u₁+u₂*cnj u₂) -
               u₁*cnj u₂*(v₁*cnj v₁+v₂*cnj v₂);
         den = u₁*cnj u₂*cnj v₁*v₂ - v₁*cnj v₂*cnj u₁*u₂
      in if den ≠ 0 then
           mk_h_line_cmat2 (Re(i*den)) (i*nom)
         else if u₁*cnj u₂ ≠ 0 then
           mk_h_line_cmat2 0 (i*u₁*cnj u₂)
         else if v₁*cnj v₂ ≠ 0 then
           mk_h_line_cmat2 0 (i*v₁*cnj v₂)
         else
           mk_h_line_cmat2 0 i)"
```

Note that the last branch is taken only if $u = v = 0$ or $\{u, v\} = \{0, \infty\}$ and returns the x-axis (this choice is quite arbitrary and is not used in the rest of our formalization, as we shall always assume that $u \neq v$, and that both points are within the unit disc). Also note that we construct unoriented circlines (although we have developed support for circline orientation considering only unoriented h-lines suffices for Poincare disc formalization).

We first show that our construction is valid i.e., that it always yields h-lines that contain two starting points.

```
lemma
assumes "u ≠ v"
shows "on_circline (h_line u v) u" "on_circline (h_line u v) v"
shows "is_h_line (h_line u v)"
```

The first claim is easily shown by direct calculation, but the second one requires a very cumbersome calculation for showing that the circline is real (i.e, that it has a negative determinant since $|B|^2 > |A|^2$). However, we avoided that by calling upon the lemma that shows that there are at least two points on this circline and by our previous results this is possible only if it is a real circline (imaginary circlines have positive determinant and the empty set of points, and circlines with the determinant zero degenerate into a single point).

All properties of h-lines propagate to h-lines constructed from two points (e.g., the h-line through u and v contains also the inverses of u and v).

The explicit line construction enables us to prove that there is a unique h-line through any given two h-points (uniqueness part was already shown earlier).

```
lemma
assumes "u ≠ v" "u ∈ unit_disc" "v ∈ unit_disc"
shows "∃! l. is_h_line l ∧ u ∈ circline_set l ∧ v ∈ circline_set l"
```

This theorem gives us many simple, but very useful corollaries. For example, if $u \neq v$, then h-line uv is the same as the h-line vu. Each h-line is the h-line constructed out of its two arbitrary different points. If an h-line contains two different points on x-axis/y-axis then it is the x-axis/y-axis.

The h-line construction is preserved by h-isometries i.e., by unit disc preserving Möbius transformations and by conjugation.

```
lemma
assumes "unit_disc_fix M" "u ∈ unit_disc" "v ∈ unit_disc" "u ≠ v"
shows "h_line (moebius_pt M u) (moebius_pt M v) =
       moebius_cl M (h_line u v)"
lemma
assumes "u ∈ unit_disc" "v ∈ unit_disc" "u ≠ v"
shows "h_line (conjugate u) (conjugate v) =
       conjugate_cl (h_line u v)"
```

Both lemmas are consequences of the h-line uniqueness. For example, since h-line uv contains both u and v, its image by any Möbius transformation contains images of u and v (since Möbius transformation preserve incidence). Those images are different (since Möbius transformations are bijections), and since we are dealing with a disc-preserving Möbius, both are in the unit disc. Therefore the image of the h-line uv must coincide by the (unique) h-line that contains images of u and v.

3.2 h-distance

Informally, the *h-distance* between the two h-points is defined as the absolute value of the logarithm of the cross ratio between those two points and the two *ideal points* i.e., the two points where the h-line connecting those two points intersects the unit disc. Formalization of this concept goes in several steps.

First we needed to show that each h-line intersects the unit circle in exactly two different points. Although this might be done in other ways, we decided to show the existence constructively, i.e., to calculate the coordinates of ideal points explicitly, as it turns out that the expressions are not too bulky. Namely, if the h-line is determined by A and B, the two intersection points are

$$\frac{B}{|B|^2}\left(-A \pm i \cdot \sqrt{|B|^2 - A^2}\right).$$

Of course, we formalize this in homogeneous coordinates, and lift it to circlines and points in $\mathbb{C}P^1$(by showing the independence of representative Hermitian matrices).

```
definition calc_ideal_point1_cvec2 :: "complex ⇒ complex ⇒ cvec2" where
   "calc_ideal_point1_cvec2 A B =
      (let discr = Re ((cmod B)^2 - (Re A)^2)
        in (B*(-A - i*sqrt(discr)), (cmod B)^2))"
definition calc_ideal_point2_cvec2 :: "complex ⇒ complex ⇒ cvec2" where
      (let discr = Re ((cmod B)^2 - (Re A)^2)
        in (B*(-A + i*sqrt(discr)), (cmod B)^2))"
definition calc_ideal_points_cmat2_cvec2 :: "cmat2 ⇒ cvec2 set" where
   "calc_ideal_points_cmat2_cvec2 H =
      (if is_h_line_cmat2 H then
          let (A, B, C, D) = H
           in {calc_ideal_point1_cvec2 A B, calc_ideal_point2_cvec2 A B }
       else
          {(-1, 1), (1, 1)})"
```

Note that although we are interested only in finding ideal points of h-lines, due to the totality of all Isabelle/HOL functions, our definition needed to cover all circlines (when the circline is not h-line, we quite arbitrarily return the points -1 and 1, as we shall always guard against such cases using the `is_h_line` assumption). Also note that as we are dealing with unoriented circlines and as the order of the two intersection points depends on circline orientation, we can only return a set of two points, and not a list (otherwise, we could not prove independence of the circline representative matrix). By using lengthy, but quite straightforward calculations we show that for every h-line its two ideal points are different and are on the intersection of that line and the unit circle.

```
lemma
assumes "is_h_line H"
shows "∀ z ∈ calc_ideal_points H. z ∈ circline_set H ∩ unit_circle_set"
      "card (calc_ideal_points H) = 2"
```

As the intersection of any two different circlines can have at most two points (as circlines are uniquely determined by their three different points), it follows that `calc_ideal_points` returns exactly the set of all intersection points of the given h-line with the unit circle (that set is denoted by `ideal_points`).

```
definition ideal_points :: "circline ⇒ cp1 set" where
   "ideal_points H = circline_set H ∩ unit_circle_set"

lemma
assumes "is_h_line H"
shows "ideal_points H = calc_ideal_points H"
```

Finally, we can hide our calculations and express results necessary for further definitions in terms of `ideal_points`.

```
lemma
assumes "is_h_line H"
obtains i1 i2 where "i1 ≠ i2" "ideal_points H = {i1, i2}"
shows "∀ z ∈ ideal_points H. z ∈ circline_set H ∩ unit_circle_set"
lemma
assumes "u ∈ unit_disc" "v ∈ unit_disc" "u ≠ v"
assumes "ideal_points (h_line u v) = {i1, i2}"
shows "i1 ≠ i2" "u ≠ i1" "u ≠ i2" "v ≠ i1" "v ≠ i2"
```

As we shall very often employ a wlog technique, and map h-lines to the x-axis it is important to know its ideal points. A very simple direct calculation shows that those are -1 and 1.

H-isometries preserve ideal points of h-lines.

```
lemma
assumes "unit_circle_fix M" "is_h_line H"
shows "ideal_points (moebius_cl M H) = (moebius_pt M) ‘ (ideal_points H)"
shows "ideal_points (conjugate_cl H) = conjugate ‘ (ideal_points H)"
```

This is easy to prove. For example, since Möbius transformations that preserve the unit circle also preserve incidence, images of original ideal points belong both to the image of h-line and to the unit circle, and are therefore ideal points of the image of h-line. As Möbius transformations are bijections, they are different, and the theorem follows as there can be at most two ideal points.

Having formalized ideal points, we can formalize h-distance in spirit of the informal definition relying on the cross-ratio. The informal definition does not explicitly state the mutual order of the ideal points in the cross ratio, and assumes that the cross-ratio is always real[5] and positive (so its real logarithm is well defined). This is all correct (when the ideal points are swapped, the cross-ratio becomes reciprocated, logarithm becomes negated, and the absolute value neutralizes the change), but it needs to be formally proved. First note that the case of $u = v$ needs to be handled separately, since there is no unique h-line through them. The definition of h-distance relies on the Hilbert's definite choice operator (THE in Isabelle/HOL) and h-distance is defined as the number such that for every choice of ideal points that number is equal to the absolute value of the logarithm of the described cross ratio. Of course, it must be shown that such number exists.

```
abbreviation Re_cross_ratio where
  "Re_cross_ratio z u v w ≡ Re (to_complex (cross_ratio z u v w))"
definition calc_h_dist :: "cp1 ⇒ cp1 ⇒ cp1 ⇒ cp1 ⇒ real" where
  "calc_h_dist u i1 v i2 = abs (ln (Re_cross_ratio u i1 v i2))"
definition h_dist_pred :: "cp1 ⇒ cp1 ⇒ real ⇒ bool" where
  "h_dist_pred u v d ⟶
     (u = v ∧ d = 0) ∨
     (u ≠ v ∧ (∀ i1 i2. ideal_points (h_line u v) = {i1, i2} ⟶
                        d = calc_h_dist u i1 v i2))"
definition h_dist :: "cp1 ⇒ cp1 ⇒ real" where
  "h_dist u v = (THE d. h_dist_pred u v d)"
```

First we shown that the described cross-ratio is always finite, positive real number.

```
lemma
```

[5] Note that such claim is slightly imprecise. In a strictly typed setting the value of the cross ratio is a number in the extended complex plane $\mathbb{C}P^1$. If different from ∞_h, then it can be converted to an ordinary complex number and we claim that its imaginary part is equal to 0. Formally we claim that `is_real (to_complex (cross_ratio u i1 v i2))`, where `is_real x ≡ Im x = 0`. For simplicity, we shall sometimes make such simplifications.

```
assumes "u ∈ unit_disc" "v ∈ unit_disc" "u ≠ v"
shows "∀ i1 i2. ideal_points (h_line u v) = {i1, i2} ⟶
              cross_ratio u i1 v i2 ≠ ∞h ∧
              is_real (to_complex (cross_ratio u i1 v i2))
              Re_cross_ratio u i1 v i2 > 0"
```

This is the first example where we employ the wlog technique and obtain a very simple proof. Using without loss of generality we can assume that $u = 0$ and v is a point on the positive part of the x-axis, within the unit disc so $0 < v < 1$. We use the following wlog lemma.

```
lemma
  assumes "u ∈ unit_disc" "v ∈ unit_disc" "u ≠ v"
  assumes "⋀ M u v. ⟦ unit_disc_fix M; u ∈ unit_disc; v ∈ unit_disc; u ≠ v;
                      P (moebius_pt M u) (moebius_pt M v) ⟧ ⟹ P u v"
  assumes "⋀ x. ⟦ is_real x; 0 < Re x; Re x < 1 ⟧ ⟹ P 0h (of_complex x)"
  shows "P u v"
```

When $u = 0$ and v is a positive real, then the cross ratio is either $(1+v)/(1-v)$ or $(1-v)/(1+v)$, depending on the order of ideal points, and in both cases it is finite, real and positive. It remains to show that the property that is being proved is preserved by unit disc preserving Möbius transformations, but it is easy since such transformations preserve h-lines and ideal points and all Möbius transformations preserve cross-ratios.

Next we can show that for every different points from the unit disc there is exactly one number that satisfies the h-distance predicate.

```
lemma
assumes "u ∈ unit_disc" "v ∈ unit_disc"
shows "∃! d. h_dist_pred u v d"
```

If the points are the same, then that number is zero, and if they are different, then there exist some pair of ideal points i_1 and i_2 and the sought number is `calc_h_dist` u i_1 v i_2. By the previous lemma that number is well defined and it is not hard to show that it remains unchanged when the ideal points are swapped.

Finally we can eliminate the need for unfolding the complicated h-distance definition by the following lemmas.

```
lemma
assumes "u ∈ unit_disc"
shows "h_dist u u = 0"
lemma
assumes "u ∈ unit_disc" "v ∈ unit_disc" "u ≠ v"
        "ideal_points (h_line u v) = {i1, i2}"
shows "h_dist u v = calc_h_dist u i1 v i2"
```

Instead of the h-distance itself, very frequently its hyperbolic cosine is analyzed (it is defined by $\cosh x = (e^x + e^{-x})/2$). It can be expressed as the average of the following two cross-ratios (shown easily by direct calculations based on the definitions of the cross-ratio and hyperbolic cosine).

```
abbreviation "cosh_h_dist u v ≡ cosh (h_dist u v)"

lemma
assumes "u ∈ unit_disc" "v ∈ unit_disc" "u ≠ v"
        "ideal_points (h_line u v) = {i₁, i₂}"
shows "cosh_h_dist u v =
       ((Re_cross_ratio u i₁ v i₂) + (Re_cross_ratio v i₁ u i₂)) / 2"
```

Since h-distance is defined by means of cross-ratio, it is easy to show that it is preserved by h-isometries i.e., by Möbius transformations that preserve the unit disc and by conjugation (we have already proved that h-lines, ideal points and cross ratio are preserved by these transforms, so the conclusion follows quite easily).

```
lemma
assumes "unit_disc_fix M" "u ∈ unit_disc" "v ∈ unit_disc"
shows "h_dist (moebius_pt M u) (moebius_pt M v) = h_dist u v"
lemma
assumes "u ∈ unit_disc" "v ∈ unit_disc"
shows "h_dist (conjugate u) (conjugate M v) = h_dist u v"
```

In some cases the h-distance can easily be calculated directly. For example, h-distance between any two points x_1 and x_2 on the segment of the x-axis within the unit disc can be calculated as

$$\left|\ln\frac{(1+x_1)(1-x_2)}{(1-x_1)(1+x_2)}\right|.$$

This further simplifies if one of the two points is zero. If one of the two points is zero, the h-distance can be calculated by

$$\left|\ln\frac{1-|x|}{1+|x|}\right|,$$

which is easily shown by employing the wlog technique and assuming that the other point is on the positive part of the x-axis (basically, by rotating the whole configuration). We use the following wlog lemma.

```
lemma
  assumes "u ∈ unit_disc" "u ≠ 0ₕ"
  assumes "⋀ φ u. ⟦ u ∈ unit_disc; u ≠ 0ₕ;
                    P (moebius_pt (moebius_rotation φ) u)⟧ ⟹ P u"
  assumes "⋀ x. ⟦ is_real x; 0 < Re x; Re x < 1 ⟧ ⟹ P (of_complex x)"
  shows "P u"
```

Explicit h-distance formula. Although we found some explicit formulas for h-distance calculations, they covered only some special cases (when the two h-points line on a h-line that is an Euclidean line). Next we show that the following formula expresses h-distance between any two h-points (note that the ideal points do not figure anymore).

$$d(u,v) = \operatorname{arccosh}\left(1 + \frac{2\cdot|u-v|^2}{(1-|u|^2)\cdot(1-|v|^2)}\right)$$

Inverse cosine is defined by $\operatorname{arccosh} x = \ln\left(x + \sqrt{x^2-1}\right)$ for $x \geq 1$, and for such x it is the inverse function of the hyperbolic cosine. Note that if u and v are in the unit disc, then the denominator is always strictly greater than zero, so the argument of arccosh is within its domain.

```
definition h_dist_formula' :: "complex ⇒ complex ⇒ real" where
  "h_dist_formula' u v =
   1 + 2 * ((cmod (u-v))^2 / ((1-(cmod u)^2) * (1-(cmod v)^2)))"
definition h_dist_formula :: "complex ⇒ complex ⇒ real" where
  "h_dist_formula u v = acosh (h_dist_formula' u v)"
```

The equivalence between the two h-distance representations is given by the following theorem.

```
lemma
assumes "u ∈ unit_disc" "v ∈ unit_disc"
shows "h_dist u v = h_dist_formula (to_complex u) (to_complex v)"
```

We prove it by employing the wlog approach and assume that $u = 0$ and that v is a real number such that $0 < v < 1$ (the case $u = v$ is discharged trivially). In that case we already know that the left-hand side is equal to $\left|\ln \frac{1-v}{1+v}\right|$ and need to prove that this is also the case with the right hand side. Since v is real, right-hand side is $\operatorname{arccosh}\left(1 + \frac{2v^2}{1-v^2}\right)$. After unfolding the inverse hyperbolic cosine definition and simplifying the expression (using the assumption that v is within the unit disc so $|v| < 1$) we get that the right-hand side is equal to $\ln \frac{1+v}{1-v}$, since v is real and positive. That value is between 0 and 1, so its logarithm is negative and its absolute value is equal to its opposite. Therefore, the left-hand side is equal the logarithm of its reciprocal, and that is equal the value of the the right-hand side (since v is real).

It still needs to be justified that the wlog assumption could have been used i.e., that the statement of the lemma is preserved by Möbius transformations that preserve the unit disc. The h-distance is defined by means of the cross-ratio and is clearly preserved, but we also need to show that the h-distance formula is also preserved. To do that we use another wlog theorem and show that the h-distance formula is preserved by Blaschke transforms and by rotations (this suffices since every unit disc preserving Möbius transform is a composition of such two transforms). We use the following wlog lemma.

```
lemma
  assumes "unit_disc_fix M"
  assumes "⋀ k. cmod k < 1 ⟹ P (blaschke k)"
  assumes "⋀ φ. P (moebius_rotation φ)"
  assumes "⋀ M1 M2. ⟦ unit_disc_fix M1; P M1;
                      unit_disc_fix M2; P M2 ⟧ ⟹ P (M1 + M2)"
  shows "P M"
```

Rotations preserve complex modulus, so they clearly preserve the h-distance formula. Showing that Blaschke factors preserve h-distance formula is done by direct calculations. We fix an arbitrary complex number k such that `cmod k < 1`, and analyze `blaschke k` transform. The h-distances between the original points and between the transformed points are given by the following formulas.

$$\operatorname{arccosh}\left(1 + 2 \cdot \frac{|u-v|^2}{(1-|u|^2)(1-|v|^2)}\right)$$

$$\operatorname{arccosh}\left(1 + 2 \cdot \frac{\left|\frac{u-k}{1-\overline{k}u} - \frac{v-k}{1-\overline{k}v}\right|^2}{\left(1 - \left|\frac{u-k}{1-\overline{k}u}\right|^2\right) \cdot \left(1 - \left|\frac{v-k}{1-\overline{k}v}\right|^2\right)}\right)$$

These two are equal since it holds that:

$$\frac{\left|\frac{u-k}{1-\overline{k}u} - \frac{v-k}{1-\overline{k}v}\right|^2}{\left(1 - \left|\frac{u-k}{1-\overline{k}u}\right|^2\right) \cdot \left(1 - \left|\frac{v-k}{1-\overline{k}v}\right|^2\right)} = \frac{\frac{|u-v|^2|1-k\overline{k}|^2}{|1-\overline{k}u|^2|1-\overline{k}v|^2}}{\frac{|1-k\overline{k}|(1-|u|^2)}{|1-\overline{k}u|^2} \cdot \frac{|1-k\overline{k}|(1-|v|^2)}{|1-\overline{k}v|^2}} =$$

$$\frac{|u-v|^2}{(1-|u|^2)\cdot(1-|v|^2)}.$$

Note that formal justification of the previous equalities is not trivial. For example, showing that

$$1 - \left|\frac{u-k}{1-\overline{k}u}\right|^2 = \left|1 - \left|\frac{u-k}{1-\overline{k}u}\right|^2\right|$$

which is one of the intermediate steps in the previous derivation requires knowing that Blaschke transform preserves the unit disc, so the subtracted value is always less than 1.

The h-distance explicit formula enables us to prove some properties of h-distance much easier than by using its cross-ratio definition. For example, it is trivial to prove that the distance is non-negative, that it is zero only if the two points are equal and that it is symmetric. Together with the following triangle inequality these properties prove that h-distance is a metric.

```
lemma
assumes "u ∈ unit_disc" "v ∈ unit_disc" "w ∈ unit_disc"
shows "h_dist u v + h_dist v w ≥ h_dist u w"
```

We have proved triangle inequality also by using the h-distance formula, and by employing the wlog principle (it can be easily justified since the h-distance is preserved by unit disc preserving Möbius transformations) and assuming that $v = 0$. It holds that `h_dist` u w is equal to

$$\operatorname{arccosh}\left(\frac{(1-|u|^2)(1-|w|^2) + 2|u-w|^2}{(1-|u|^2)(1-|w|^2)}\right)$$

The sum `h_dist` u 0 `+` `h_dist` 0 w is equal to

$$\operatorname{arccosh}\left(1 + \frac{2|u|^2}{(1-|u|^2)}\right) + \operatorname{arccosh}\left(1 + \frac{2|w|^2}{(1-|w|^2)}\right)$$

When the addition formula

$$\operatorname{arccosh} x + \operatorname{arccosh} y = \operatorname{arccosh} xy + \sqrt{(x^2-1)(y^2-1)}$$

is applied and the obtained expression is simplified, the previous formula is transformed to

$$\operatorname{arccosh}\left(\frac{(1+|u|^2)(1+|w|^2)+4|u||w|}{(1-|u|^2)(1-|w|^2)}\right)$$

Since arccosh is monotone increasing function, it suffices to show that

$$(1-|u|^2)(1-|w|^2)+2|u-w|^2 \leq (1+|u|^2)(1+|w|^2)+4|u||w|,$$

i.e., that

$$|u-w|^2 \leq (|u|+|w|)^2$$

which holds by a triangle inequality $|u-w| \leq |u|+|w|$ for real numbers.

Existence of an h-isometry (unit disc preserving Möbius transform eventually composed by conjugation) between two pairs of points shows that they are equidistant. The converse also holds i.e., the group of Möbius transforms that preserve the unit disc is rich enough to map every pair of points to another pair of points with the same h-distance.

```
lemma
assumes "{u, v, u', v'} ⊆ unit_disc"
assumes "h_dist u v = h_dist u' v'"
shows "∃ M. unit_disc_fix M ∧ moebius_pt M u = u' ∧ moebius_pt M v = v'"
```

We first prove a lemma showing that for each real number d there is exactly one point on the positive x-axis such that h-distance between 0 and that point is d (this is done either by showing that the restriction of the distance function to positive x-axis that is $\ln\frac{1+x}{1-x}$ is bijection between positive x-axis and $\mathbb{R}^+$ or by using the existence and uniqueness of cross-ratio).

If $u' = v'$, the theorem holds trivially. Otherwise, wlog we assume that $u' = 0$ and that v' is on the positive x-axis. Then we find a unit disc preserving Möbius that maps u to 0 and v to the positive x-axis (it exists by our previous results and is a composition of a Blaschke transform and a rotation). By the previous uniqueness lemma, the image of v must be equal to v'.

3.3 h-circles

In every geometry circles consist of points that are at the same distance from the center.

```
definition h_circle :: "cp1 ⇒ real ⇒ cp1 set" where
  "h_circle u r = {z. z ∈ unit_disc ∧ h_dist u z = r}"
```

Note that by our definition of distance there could theoretically exist some points out of the disc that have distance r from the center u, but our definition of h-circle explicitly excluded them.

The central result is that h-circles are represented by Euclidean circles in the Poincaré model. An h-circle centered at u with the radius r is represented by an Euclidean circle centered at

$$u_e = \frac{u}{(1-|u|^2)\frac{\cosh r - 1}{2} + 1}$$

and radius

$$r_e = \frac{(1-|u|^2)\sqrt{\frac{\cosh r-1}{2}\cdot\frac{\cosh r+1}{2}}}{(1-|u|^2)\frac{\cosh r-1}{2}+1} = \frac{(1-|u|^2)\sinh r}{(1-|u|^2)(\cosh r - 1) + 2}$$

This is formalized by the following definition.

```
definition h_circle_euclidean :: "cp1 ⇒ real ⇒ (complex × real)" where
  "h_circle_euclidean u r =
     (let R = (cosh r - 1) / 2;
          u' = to_complex u;
          cu = 1 - (cmod u')²;
          k = cu * R + 1
       in (u' / k, cu * sqrt(R * (R + 1)) / k))"
```

First we show that previous formulas describe an Euclidean circle that always has a positive radius and is always fully completely situated within the unit disc.

```
lemma
assumes "r > 0" "z ∈ unit_disc" "(z_e, r_e) = h_circle_euclidean z r"
shows "cmod z_e < 1" "r_e > 0" "∀ x ∈ circle z_e r_e. of_complex x ∈ unit_disc"
```

Let us demonstrate this. Let $R = \frac{\cosh r - 1}{2}$ and $U = 1 - |u|^2$. If $r > 0$ then $\cosh r > 1$ and $R > 0$. Since u is in the unit disc it holds that $U > 0$, so the denominator $UR + 1$ is strictly greater than one, and the center of the circle obtained by dividing the h-center by this denominator must be in the unit disc. Since $r > 0$ it holds that $\sinh r > 0$, so the nominator $U \sinh r$ and the Euclidean radius are also strictly positive. It remains to show that every point in the circle is within the unit disc. If z is an arbitrary point on the circle, then by a triangle inequality it holds that $|z| \le |u_e| + r_e$. We shall show that the right hand side is always strictly less than 1. By the inequality of geometric and arithmetic mean it holds that $\sqrt{R(R+1)} < \frac{R+(R+1)}{2}$, i.e., that $\sqrt{R(R+1)} - R < 1/2$. Since u is within the unit disc, it holds that $1 + |u| < 2$, so $(1+|u|)(\sqrt{R(R+1)} - R) < 1$. Therefore, $U(\sqrt{R(R+1)} - R) < 1 - |u|$ and $|u| + U\sqrt{R(R+1)} < 1 + RU$. Since $1 + RU > 1$ and $|z_e| = |u|/(1+RU)$ and $r_e = (U\sqrt{R(R+1)})/(1+RU)$ it holds that $|u_e| + r_e < 1$.

The connection between the points on the h-circle and its corresponding Euclidean circle is given by the following lemma.

```
lemma
assumes "u ∈ unit_disc" "r > 0"
shows "let (u_e, r_e) = h_circle_euclidean u r
       in h_circle u r = of_complex ` (circle u_e r_e) "
```

Let z be an arbitrary point. By the h-distance formula, it belongs to the h-circle iff $|z| < 1$ and $1 + \frac{2|z-u|^2}{(1-|z|^2)(1-|u|^2)} = r$. Let $R = \frac{\cosh r - 1}{2}$ and $U = 1 - |u|^2$. The following equivalences hold.

$$
\begin{aligned}
&|z| < 1 \wedge 1 + \frac{2|z-u|^2}{(1-|z|^2)(1-|u|^2)} = r \longleftrightarrow \\
&|z| < 1 \wedge |z-u|^2 = R \cdot (1-|z|^2) \cdot U \longleftrightarrow \\
&|z| < 1 \wedge (z-u)(\overline{z}-\overline{u}) + R \cdot U \cdot z\overline{z} = RU \longleftrightarrow \\
&|z| < 1 \wedge z\overline{z} - \frac{z\overline{u}}{1+RU} - \frac{\overline{z}u}{1+RU} + \frac{u\overline{u}}{1+RU} = \frac{RU}{1+RU} \longleftrightarrow \\
&|z| < 1 \wedge \left|z - \frac{u}{1+RU}\right|^2 - \frac{|u|^2}{(1+RU)^2} + \frac{|u|^2}{1+RU} = \frac{RU}{1+RU} \longleftrightarrow \\
&|z| < 1 \wedge \left|z - \frac{u}{1+RU}\right|^2 = \frac{U^2 \cdot R(R+1)}{(1+RU)^2} \\
&|z| < 1 \wedge |z - z_e|^2 = r_e^2
\end{aligned}
$$

By the previous lemma the whole circle is within the unit disc so $|z - z_e| = r_e^2$ implies that $|z| < 1$, and the equivalence is proved.

The results about circles that were just proved give us a very simple proof of the following theorem about h-isometries, that happens to be a crucial step in proving that the Poincaré disc satisfies some of the Tarski axioms (e.g., the five-segment axiom). For every pair of triangles such that its three pairs of sides are pairwise equal there is an h-isometry (a unit disc preserving Möbius transform, eventually composed with a conjugation) that maps one triangle onto the other.

```
lemma
assumes
  "u ∈ unit_disc" "v ∈ unit_disc" "w ∈ unit_disc" and
  "u' ∈ unit_disc" "v' ∈ unit_disc" "w' ∈ unit_disc" and
  "h_dist u v = h_dist u' v'"
  "h_dist v w = h_dist v' w'"
  "h_dist u w = h_dist u' w'"
shows
  "∃ M. unit_disc_fix_fun M ∧ M u = u' ∧ M v = v' ∧ M w = w'"
```

If any two points of the first triangles are the same, then the proof is trivial (and reduces to a previously proved lemmas about congruent segments). Otherwise, wlog we can assume that $u' = 0$ and that v' is on the positive x-axis. Let M be a Möbius transform that preserves the unit disc and maps u to u' and v to v' (it exists since the h-distance between u and v is equal to the h-distance between u' and v').

Then w must be an intersection of the h-circle centered in u with radius equal to h-distance d_{uw} from u to w and the h-circle centered in v with radius equal to h-distance d_{vw} from v to w. Also, since Möbius transforms preserve distances, the image of $M(w)$ must also be an intersection of those two h-circles. Now, both these h-circles are Euclidean circles, with centers on the x-axis and positive radii. As the circles intersect and cannot be the same (since v is not 0), their Euclidean centers cannot be in the same point. Therefore, by a simple analysis of their Euclidean equations it holds that they intersect in a two points that are mutually conjugate. So $M(w)$ is either equal to w' or is its conjugate. Depending on this we either just use M or compose it with a conjugation (that fixes both u and v, since they are on the x-axis) and obtain the desired transformation.

3.4 h-between

There are several equivalent ways to define h-betweenness of three points. In Poincaré model h-lines can be Euclidean circles and there is no notion of betweenness of three points on a circle (since two points divide circles into two different arc, and the third point is between the two on one of those two arcs). However, when two points are fixed, by means of cross ratio we can decide if some other two given points are on the same or on different of the two arcs determined by the fixed points. Namely, if they are on a same arc the cross ratio is positive, and otherwise it is negative (for simplicity we assume that all points are different). Since four points are on the same circline, their cross-ratio must be real. Also note that the inverse point of v is not in the unit disc and therefore is always different from the other three points. This analysis can be used to define h-betweenness. Namely, if two outer points are fixed, the inner point is not between them iff it is on the same arc as the arc of the h-line that goes out of the unit disc. We know that the inverse of that point is not in the unit disc, so the point v is between u and w if the cross-ratio between the pairs u and w and v and inverse of v is real and negative. Since Tarski considers non-strict betweenness, we extend the definition by two special cases (there must be special cases since when $u = v = w$, the cross-ratio is not well defined).

```
definition h_betw :: "cp1 ⇒ cp1 ⇒ cp1 ⇒ bool" where
  "h_betw u v w ⟷
   u = v ∨ v = w ∨
   (let cr = cross_ratio u v w (inversion v)
     in is_real (to_complex cr) ∧ Re (to_complex cr) < 0)"
```

Some basic properties of the h-betweeness follow directly from this definition. For example, for three points `u`, `v` and `w` within the unit disc `h_betw u u v` and `h_betw u v v` follow directly from the two special cases, while `h_betw u v u` implies `u = v`, since otherwise the cross-ratio would be 1, and not negative. By exchanging points `u` and `w` cross-ratio is reciprocated, so it sign does not change, so `h_betw u v w` iff `h_betw w v u`. All these simple lemmas are easy to prove by properties of the cross-ratio. However, attention must be put on discharging assumptions that no three points are equal, so cross-ratio is well-defined. Note that some of these lemmas might have simpler proofs with some other characterizations of betweenness.

Since Möbius transforms that preserve unit disc preserve cross-ratio and inverse points, they preserve betweeness. The same hold for conjugation and, therefore, for all h-isometries.

```
lemma
assumes "unit_disc_fix M" "u ∈ unit_disc" "v ∈ unit_disc" "w ∈ unit_disc"
shows
  "h_betw (moebius_pt M u) (moebius_pt M v) (moebius_pt M w) ⟷ h_betw u v w"
  "h_betw (conjugate M u) (conjugate M v) (conjugate M w) ⟷ h_betw u v w"
```

If the three points lie on an h-line that is a Euclidean line (e.g., if one of the points is zero), h-betweenness can be characterized in the following way (much simpler than the definition).

```
lemma
assumes "u ∈ unit_disc" "v ∈ unit_disc" "u ≠ 0ₕ" "v ≠ 0ₕ"
shows "h_betw u 0ₕ v ⟷ (∃ k < 0. to_complex u = k * to_complex v)"
      "h_betw 0ₕ u v ⟷ (let u' = to_complex u; v' = to_complex v
                          in arg u' = arg v' ∧ cmod u' ≤ cmod v')"
```

Both these characterizations are proved by wlog assumption that one of the two points is on the positive part of the x-axis and proving the betweenness characterization for three real numbers (for example 0 is between x and y iff $x \cdot y \leq 0$, and x is between 0 and y if $0 \leq x \leq y$ or $y \leq x \leq 0$). It is easily shown that rotations preserve all relevant expressions, so the wlog assumption is justified.

Three points can be in an h-between relation only when they are h-collinear — since the cross ratio is real, they all must lie on a single circline. Namely, h-line through u and v also contains u v and `inversion v`, so by the uniqueness of circline through three different points, that circline is an h-line and the points are collinear. The converse is also valid in some sense, since for three h-collinear points at least one of the three possible h-betweeness relations must hold.

```
lemma
assumes "u ∈ unit_disc" "v ∈ unit_disc" "w ∈ unit_disc"
shows "h_collinear {u, v, w} ⟷
       h_betw u v w ∨ h_betw u w v ∨ h_betw v u w"
```

After discharging degenerate cases, the lemma is proved by assuming that $u = 0$ and that v is a positive real number, and by using the properties of the betweenness relation between points on Euclidean lines (i.e., for the x-axis).

Another possible definition of the h-betweenness relation is given in terms of h-distances between pairs of points. We prove it as a characterization equivalent to our cross-ratio based definition (this could have been taken as the h-betweenness definition).

```
lemma
assumes "u ∈ unit_disc" "v ∈ unit_disc" "w ∈ unit_disc"
shows "h_betw u v w ⟷ h_dist u v + h_dist v w = h_dist u w"
```

After eliminating degenerate cases, we prove the theorem by a wlog assumption that $v = 0$ and that w is a point on the positive part of x-axis. By previous results we know that `h_betw` u v w holds only if `u` is a real negative number. We need to show the same for the sum of distances. We show that `h_dist` u `0 + h_dist 0` w `= h_dist` u w iff u is real and negative. By the same calculations as in the proof of triangle inequality for h-distance we show that previous is equivalent to $|u - w|^2 = (|u| + |w|)^2$. Since $\operatorname{Im} w = 0$, this is equivalent to $(\operatorname{Re} u - \operatorname{Re} w)^2 + (\operatorname{Im} u)^2 = (\operatorname{Re} u)^2 + (\operatorname{Im} u)^2 + 2\sqrt{((\operatorname{Re} u)^2 + (\operatorname{Im} u)^2)(\operatorname{Re} w)^2} + (\operatorname{Re} w)^2$, i.e., to $|\operatorname{Re} w|\sqrt{(\operatorname{Re} u)^2 + (\operatorname{Im} u)^2} + \operatorname{Re} u \cdot \operatorname{Re} w = 0$. Hence $\sqrt{(\operatorname{Re} u)^2 + (\operatorname{Im} u)^2} = |\operatorname{Re} u|$ and therefore $\operatorname{Im} u = 0$ and u is also real. Plugging $\operatorname{Im} u = 0$ into the previous equality yields that $|\operatorname{Re} w| \cdot |\operatorname{Re} u| + \operatorname{Re} w \cdot \operatorname{Re} u = 0$. Since $|\operatorname{Re} w| \cdot |\operatorname{Re} u|$ is positive, $\operatorname{Re} w \cdot \operatorname{Re} u$ must be negative.

Note that some properties of h-betweenness are very easily proved by using the previous characterization.

3.5 Intersection of h-lines with the axes

Pasch's axiom proves existence of the intersection of two circlines (and its position). The wlog technique enables us to simplify calculations by assuming that one of the two circlines is the x-axis. In some other cases (e.g., when considering perpendicularity) we shall also consider intersections with the y-axis. Therefore in this section we shall introduce some techniques for reasoning about such intersections.

First we define the condition that an h-line intersects the x-axis. Since the equation of x-axis is $z = \overline{z}$, the intersection point z of an h-line determined by A and B and the x-axis is real and it must satisfy $Az^2 + (B + \overline{B})z + A = 0$, i.e., $Az^2 + 2 \cdot \operatorname{Re} B \cdot z + A = 0$. If $A = 0$, then the intersection exists and is equal to 0. Otherwise the intersection exists iff the discriminant is non-negative. If it is equal to zero, then $\operatorname{Re} B = A$ or $\operatorname{Re} B = -A$, so $z = 1$ or -1 and none of the intersection points is within the unit disc. If it is strictly positive, then there are two different intersection points. By Vieta's formula it holds that their product is $A/A = 1$. Therefore, one of the two intersection points must be within the unit disc.

This inspires the following definition.

```
definition intersects_x_axis_cmat2 :: "cmat2 ⇒ bool" where
  "intersects_x_axis_cmat2 H =
   (let (A, B, C, D) = H in A = 0 ∨ (Re B)² > (Re A)²)"
lift_definition intersects_x_axis :: "circline ⇒ bool" is ...
```

The following lemma, whose proof is based on the previous analysis, justifies such definition.

```
lemma
assumes "is_h_line H"
shows "(∃ x ∈ unit_disc. x ∈ circline_set H ∩ circline_set x_axis) ⟷
       intersects_x_axis H"
```

Existence of an intersection with the y-axis is characterized quite similarly.

```
definition intersects_y_axis_cmat2 :: "cmat2 ⇒ bool" where
   "intersects_y_axis_cmat2 H =
    (let (A, B, C, D) = H in A = 0 ∨ (Im B)^2 > (Re A)^2)"
lift_definition intersects_y_axis :: "circline ⇒ bool" is ...
lemma
assumes "is_h_line H"
shows "(∃ y ∈ unit_disc. y ∈ circline_set H ∩ circline_set y_axis) ⟷
       intersects_y_axis H"
```

Although this can be proved directly, it is proved more easily by rotating the configuration 90 degrees in the negative direction.

The intersection point within the unit disc can be calculated by the following formula.

$$x = \frac{-\operatorname{Re} B}{A} + \frac{\operatorname{sgn}(\operatorname{Re} B) \cdot \sqrt{(\operatorname{Re} B)^2 - A^2}}{A}$$

```
definition calc_x_axis_intersection_cmat2_cvec2 :: "cmat2 ⇒ cvec2" where
   "calc_x_axis_intersection_cmat2_cvec2 H =
      (let (A, B, C, D) = H
        in if A = 0 then (0, 1)
           else (-Re(B) + sgn (Re B) * sqrt((Re B)^2 - (Re A)^2), A))"
lift_definition calc_x_axis_intersection :: "circline ⇒ cp1" is ...
```

It is characterized by the following lemma.

```
lemma
assumes "is_h_line H" "intersects_x_axis H"
shows "calc_x_axis_intersection H ∈ unit_disc"
      "calc_x_axis_intersection H ∈ circline_set H ∩ circline_set x_axis"
```

The only interesting part of the proof is that the intersection point is in the unit disc. If $\operatorname{Re}\ A = A = 0$, that is trivial. Otherwise $\operatorname{Re} B \neq 0$ and it holds that $\sqrt{(\operatorname{Re}\ B)^2 - A^2} < |B| = \operatorname{sgn}(\operatorname{Re}\ B) \cdot \operatorname{Re}\ B$. Multiplying this by $2 \cdot \sqrt{(\operatorname{Re}\ B)^2 - A^2}$, rearranging the terms and using the fact that $\operatorname{sgn}(\operatorname{Re}\ B)^2 = 1$ gives

$$\left(\frac{-\operatorname{Re}\ B + \operatorname{sgn}(\operatorname{Re}\ B) \cdot \sqrt{(\operatorname{Re}\ B)^2 - A^2}}{A}\right)^2 < 1.$$

We shall also be interested if the intersection point is on the positive part of the x-axis/y-axis.

```
definition intersects_x_axis_positive_cmat2 :: "cmat2 ⇒ bool" where
   "intersects_x_axis_positive_cmat2 H =
       (let (A, B, C, D) = H in Re A ≠ 0 ∧ Re B / Re A < −1)"
lift_definition intersects_x_axis_positive :: "circline ⇒ bool" is ...
```

```
definition intersects_y_axis_positive_cmat2 :: "cmat2 ⇒ bool" where
   "intersects_x_axis_positive_cmat2 H =
       (let (A, B, C, D) = H in Re A ≠ 0 ∧ Im B / Re A < −1)"
lift_definition intersects_y_axis_positive :: "circline ⇒ bool" is ...
```

Characterization is given by the following lemma (analogous one is proved for the y-axis).

```
lemma
assumes "is_h_line H" "H ≠ x_axis"
shows "intersects_x_axis_positive H ⟷
       (∃ x. x ∈ unit_disc ∧ x ∈ circline_set H ∩ positive_x_axis)"
```

The intersection point is, of course, again given by `calc_x_axis_intersection`. If $\mathrm{Re}\, A = A \neq 0$ and $(\mathrm{Re}\, B)^2 > A^2$ (so that the square root is well-defined and there is an intersection in the disc), it holds that

$$\left|\frac{\mathrm{sgn}\,(\mathrm{Re}\, B) \cdot \sqrt{\mathrm{Re}(B)^2 - A^2}}{A}\right| < \left|\frac{\mathrm{sgn}\,(\mathrm{Re}\, B) \cdot |\,\mathrm{Re}\, B|}{A}\right| = \left|\frac{\mathrm{Re}\, B}{A}\right| = \left|\frac{-\,\mathrm{Re}\, B}{A}\right|.$$

Therefore the sign of x depends only on the sign of $\frac{-\mathrm{Re}\, B}{A}$. Since $(\mathrm{Re}\, B)^2 > A^2$, if $\frac{-\mathrm{Re}\, B}{A}$ is positive then it is greater than 1, and if it is negative, then it is less than 1. Therefore $\frac{-\mathrm{Re}\, B}{A} > 1$ is the condition equivalent to that x is positive.

Finally, the position of the intersection depends on the value of $\frac{-\mathrm{Re}\, B}{A}$. The smaller that number, the larger the x-coordinate of the intersection. We introduce the function that compares two circlines and tells if the x-axis intersection of the first one is more towards the edge of the unit disc than the x-axis intersection of the other one.

```
definition x_axis_intersection_outward_cmat2 :: "cmat2 ⇒ cmat2 ⇒ bool" where
   "x_axis_intersection_outward_cmat2 H1 H2 =
      (let (A1, B1, C1, D1) = H1; (A2, B2, C2, D2) = H2
        in -Re B1/Re A1 ≤ -Re B2/Re A2)"
lift_definition x_axis_intersection_outward :: "circline ⇒ circline ⇒ bool" is ...
```

The characterization of this function is given by the following lemma.

```
lemma
assumes "is_h_line H1" "is_h_line H2"
assumes "intersects_x_axis_positive H1" "intersects_x_axis_positive H2"
assumes "x_axis_intersection_outward H1 H2"
shows "Re (to_complex (calc_x_axis_intersection H1)) ≥
       Re (to_complex (calc_x_axis_intersection H2))"
```

For $\frac{\mathrm{Re}\, B}{A} < -1$ the position of the x-axis intersection point within the disc is given by $\frac{-\mathrm{Re}\, B}{A} - \sqrt{(\frac{-\mathrm{Re}\, B}{A})^2 - 1}$ (easily shown by case analysis on the sign of $\mathrm{Re}\, B$). The lemma follows from the monotonicity of the real function $x - \sqrt{x^2 - 1}$ on the interval $(1, +\infty)$.

The next lemma about the h-betweenness will be one of the crucial steps in our proof of the Pasch's axiom in the Poincaré model. We want to characterize the h-betweenness of the intersection point of two h-lines wrt. the points determining those h-lines. Our wlog technique will enable us to consider only intersections of h-lines with the x-axis. The following lemma says that the intersection point of the h-line determined by points u and v and the x-axis is between u and v, then u and v are in the opposite half-planes (one must be in the upper, and the other one in the lower half-plane).

```
lemma
assumes "u ∈ unit_disc" "v ∈ unit_disc" "z ∈ unit_disc" "u ≠ v"
assumes "u ∉ circline_set x_axis" "v ∉ circline_set x_axis"
assumes "z ∈ circline_set (h_line u v) ∩ circline_set x_axis"
shows "h_betw u z v ⟷ arg (to_complex u) * arg (to_complex v) < 0"
```

The central idea of the proof is to assume wlog that the intersection point z is zero. Then u, v and 0 are h-collinear and collinear (as their h-line contains 0 it must be a line). We recall that point 0 is h-between u and v iff there is a negative real number k such that $u = k \cdot v$. If there is such k, then $\arg\ u = \arg\ (-v)$, and since u and v are not on the x-axis, $\arg\ u$ and $\arg\ v$ have the opposite sign. For the other direction, since u, v and 0 are collinear, u and v can be represented by $u = r_u \cdot e^{i\phi}$ and $v = r_v \cdot e^{i\phi}$, for some real numbers $r_u \neq 0$, $r_v \neq 0$ and ϕ. But then if $\arg\ u \cdot \arg\ v < 0$, it must hold that $r_u \cdot r_v < 0$ and that $\frac{r_u}{r_v} < 0$. Since $u = \frac{r_u}{r_v} v$, we have found a negative number k such that $u = k \cdot v$, so the h-betweenness holds.

The wlog assumption must be justified. The key argument is that there is a Blaschke transform that maps z to 0, but of a special kind, since z is on the x-axis (in the unit disc). We use the following wlog lemma.

```
lemma
  assumes "u ∈ unit_disc" "is_real (to_complex u)"
  assumes "⋀ a u. ⟦ u ∈ unit_disc; is_real a; cmod a < 1;
                    P (moebius_pt (blaschke a) u)⟧ ⟹ P u"
  assumes "P 0ₕ"
  shows "P u"
```

Blaschke transforms determined by a real parameter a preserve the x-axis (they map a to 0, $1/\overline{a}$ to ∞ and 0 to $-1/\overline{a} = -1/a$, so three different points on the x-axis stay on the x-axis, so it must be fixed). Also, such Blaschke transforms preserve the sign of the imaginary part.

```
lemma
assumes "is_real a" "cmod a < 1" "z ≠ ∞ₕ" "z ≠ inversion (of_complex a)"
shows "sgn (Im (to_complex (moebius_pt (blaschke a) z))) =
       sgn (Im (to_complex z))"
```

Indeed, image of a point z is $\frac{z-a}{1-\overline{a}z}$, and since $a = \overline{a}$, its imaginary part is $\frac{(1-a^2)(Im\ z)}{|1-az|^2}$, and its sign is the same as the sign of $Im\ z$. For points not on the x-axis the sign of the imaginary part matches the sign of the argument, so such Blaschke transforms preserve the sign of the argument of non-zero points. Blaschke transforms are Möbius transforms that preserve the unit disc, so they preserve the h-betweenness. Since all the elements of the h-betweenness characterization that we prove are preserved (unit disc, x-axis, h-betweenness, and h-lines), the w-log assumption is justified.

4 Proving the Tarski's axioms

Tarski's axioms are given in a locale that is interpreted by specifying the type of points, congruence and betweenness relation over that type. We shall use the `h_point` type defined before and need to define those relations.

H-congruence between pairs of points can be defined in terms of h-distance function.

```
definition h_cong :: "cp1 ⇒ cp1 ⇒ cp1 ⇒ cp1 ⇒ bool" where
  "h_cong u v u' v' ⟷ h_dist u v = h_dist u' v'"
```

Equivalently, congruence could be defined by requiring an h-isometry that maps the first pair onto the second one.

H-congruence can then be lifted to the `h-point` type (we denote it by $\equiv_h$).

```
lift_definition ≡h :: "h_point ⇒ h_point ⇒ h_point ⇒ h_point ⇒ bool" is h_cong
```

Similarly, h-betweenness can also easily be lifted to the h-point type (and we denote it by $\mathfrak{B}_h$).

```
lift_definition 𝔅h :: "h_point ⇒ h_point ⇒ h_point ⇒ bool" is h_betw
```

In that way we obtain elements that interpret the Tarski axioms locale and need to prove axioms. Each axiom formulated over the `h_point` type is proved by transfering it to the `cp1` type and proving it in its unlifted form. Therefore, in this section we shall show unlifted variants of axioms.

Congruence axioms. For example, the axiom

```
lemma
fixes u v :: h_point
shows "u v ≡h v u"
```

is proved by transfering it to its unlifted counterpart

```
lemma
fixes u v :: cp1
assumes "{u, v} ⊆ unit_disc"
shows "h_cong u v v u"
```

This lemma is easily proved by our earlier results on h-distance. The same hold for the following two h-congruence axioms.

```
lemma
assumes "{u, v, u1, v1, u2, v2} ⊆ unit_disc"
assumes "h_cong u v u1 v1" "h_cong u v u2 v2"
shows "h_cong u1 v1 u2 v2"
```

```
lemma
assumes "{u, v, w} ⊆ unit_disc"
assumes "h_cong u v w w"
shows "u = v"
```

Segment construction axiom.

```
lemma
assumes "{u, v, a, b} ⊆ unit_disc"
shows "∃ w. w ∈ unit_disc ∧ h_between u v w ∧ h_cong v w = h_dist a b"
```

If $u = v$, wlog we can assume that $u = v = 0$. Then there exists a point w on the x-axis such that its h-distance from 0 is equal to the given h-distance from a to b.

If $u \neq v$, wlog we can assume that $v = 0$ and that u is a point on the positive x-axis (within the unit disc). Then there exists a point w on the negative part of x-axis (within the unit disc) such that its h-distance from 0 is equal to the given h-distance from a to b.

Wlog assumptions are easily justified by the fact that Möbius transformations that preserve the unit disc preserve h-betweenness and h-distance.

Five segment axiom.

```
lemma
assumes "{x, y, z, u} ⊆ unit_disc" "{x', y', z', u'} ⊆ unit_disc"
        "x ≠ y" "h_betw x y z" "h_betw x' y' z'" and
        "h_cong x y x' y'" "h_cong x u x' u'"
        "h_cong y u y' u'" "h_cong y z y' z'"
shows "h_cong z u z' u'"
```

From our previous results we know that there is an h-isometry M mapping x, y, and u to x', y', and u', respectively. We show that it maps z to z', by using the fact that there is an unique point on a ray that is on a given h-distance from the ray origin. We consider the ray $x'y'$. By assumption it contains z'. Also, since h-isometries preserve betweeness it must contain $M\ z$ (since its pre-image xy by assumption contains z). Also we know that the h-distance between x' and z' is the same as the h-distance between x' and Mz (as it is the same as the h-distance between z and z and since h-isometries preserve h-distance). Therefore M maps z to z' and the theorem holds as h-isometries preserve distance.

Identity of betweenness. Identity of betweenness has already been proved.

```
lemma
assumes "{u, v} ⊆ unit_disc"
assumes "h_betw u v u"
shows "u = v"
```

Pasch's axiom. Pasch's axiom is the most difficult lemma to prove for the Poincaré model. The following lemma expresses its non-degenerate case.

```
lemma
assumes "{x, y, z, u, v} ⊆ unit_disc"
assumes "distinct [x, y, z, u, v]" "¬ h_collinear x, y, z"
assumes "h_betw x u z" "h_betw y v z"
shows "∃ a. a ∈ unit_disc ∧ h_betw u a y ∧ h_betw x a v"
```



Fig. 7: Pasch's axiom – wlog position

The central idea of the proof is, of course, to use the wlog approach. However, the choice of points must be such that the intersection point that we claim to exist lies on the x-axis. One such choice leading to the proof is to assume that $x = 0$ and that v is a positive real number. Since u is h-between 0 and z, by our previously proved characterizations of h-betweenness we know that $\arg u = \arg z$ and that $|u| \leq |z|$. Since y, z, u and v are different from $x = 0$, we can switch to polar coordinates and assume that $u = |u|e^{i\phi}$, $z = |z|e^{i\phi}$ and $y = |y|e^{i\theta}$. Point v is between y and z and is on the positive x-axis and within the unit disc. Since neither y nor z can be on the x-axis (otherwise x, y and z would be collinear), by our earlier characterization of the h-betweenness of the x-axis intersection we know that y and z must be in different half-planes and that their arguments θ and ϕ must have opposite signs. We want to show that the h-line yu also intersects the x-axis, by knowing that the h-line yz intersects the x-axis. We have seen that the existence of that intersection depends of whether it holds that $(\operatorname{Re} B)^2 > A^2$, where A and B are matrix coefficients that determine the h-line. Knowing the polar forms for y and u, and matrix construction for h-lines, after simple calculations we show that coefficients A_{yu} and B_{yu} for the h-line yu can be expressed by

$$A_{yu} = 2|y||u|\sin(\phi - \theta), \quad B_{yu} = i \cdot (|u|e^{i\phi}(1 + |y|^2) - |y|e^{i\theta}(1 + |u|^2)).$$

Therefore

$$\operatorname{Re} B_{yu} = |y|(1 + |u|^2)\sin\theta - |u|(1 + |y|^2)\sin\phi,$$

and

$$\frac{\operatorname{Re} B_{yu}}{A_{yu}} = \frac{\sin\theta}{2\sin(\phi - \theta)} \cdot \left(\frac{1}{|u|} + |u|\right) - \frac{\sin\phi}{2\sin(\phi - \theta)} \cdot \left(\frac{1}{|y|} + |y|\right).$$

Note that $\sin(\phi - \theta)$ cannot be zero as $x = 0$, y and z are not collinear. Also, neither $\sin\phi$ nor $\sin\theta$ can be zero, since y and z are not on the x-axis.

Very similar formulas hold for the h-line yz (obtained by replacing all occurrences of u by z).

We show that

$$\frac{\operatorname{Re} B_{yu}}{A_{yu}} \leq \frac{\operatorname{Re} B_{yz}}{A_{yz}} < -1.$$

The latter inequality holds as we know that the h-line yz intersects the x-axis in a positive point v.

After substituting the derived formulas and canceling common terms, proving the first inequality reduces to proving that

$$\frac{\sin\theta}{\sin(\phi-\theta)}\left(\frac{1}{|u|}+|u|\right) \leq \frac{\sin\theta}{\sin(\phi-\theta)}\left(\frac{1}{|z|}+|z|\right).$$

Since the real function $\frac{1}{x}+x$ is decreasing on the interval $(0,1)$, and since $|u| \leq |z|$, this holds iff $\frac{\sin\theta}{\sin(\phi-\theta)}$ is negative. We show that this is the case. If $\phi > 0$, then it holds that $\sin\phi > 0$, $\theta < 0$, and $\sin\theta < 0$. Then it holds that $\operatorname{Re} B_{yz} = |y|(1+|z|^2)\sin\theta - |z|(1+|y|^2)\sin\phi < 0$. Since it holds that $\frac{\operatorname{Re} B_{yz}}{A_{yz}} < -1$, it must be that $A_{yz} > 0$. Since $A_{yz} = 2|y||z|\sin(\phi-\theta)$, it must hold that $\sin(\phi-\theta) > 0$, and $\frac{\sin\theta}{\sin(\phi-\theta)}$ is negative. The case $\phi < 0$ is proved analogously (then it holds that $\sin\theta > 0$, but $\sin(\phi-\theta) < 0$)). Therefore, we know that the line yu also intersects the x-axis in a positive point and we know that that point is more inward the center of the disc then the point v. Therefore, that point is between 0 and v. Finally, we know that the point is also between y and u, as their arguments ϕ and θ have opposite signs (in those proves we rely on our previously derived characterizations of h-betweenness).

Degenerate cases need to be proved separately. If some of the points x, y, z, u or v coincide, or if x, y and z are h-collinear, the sought point a also coincides with some of them, and the axiom is easily proved due to the non-strictness of betweenness.

Lower dimension axiom.

```
lemma lower_dimension_axiom:
shows "∃ a ∈ unit_disc. ∃ b ∈ unit_disc. ∃ c ∈ unit_disc.
        ¬ h_betw a b c ∧ ¬ h_betw b c a ∧ ¬ h_betw c a b"
```

Lower dimension is proved very easily by showing that 0, 1/2, and $i/2$ are not h-collinear (since calculations show that $i/2$ does not lie on the x-axis which the h-line between 0 and 1/2) and so none of the h-betweenness can hold (as h-betweenness implies h-collinearity).

Upper dimension axiom.

```
lemma
assumes "{x, y, z, u, v} ⊆ unit_disc" "u ≠ v"
assumes "h_cong x u x v" "h_cong y u y v" "h_cong z u z v"
shows "h_betw x y z ∨ h_betw y z x ∨ h_betw z x y"
```



Fig. 8: Upper dimension axiom – wlog position

This axiom is easily proved by using wlog reasoning and our previous results on h-circles. If any two of the points x, y, and z are the same, the statement trivially holds. Otherwise, wlog we can assume that $x = 0$ and that y is a point on the positive x-axis. Since $u \neq v$, h-distance between u and x, h-distance between u and y, and h-distance between u and z are strictly positive (h-distance is always non-negative and if it would be zero, than it would hold that $u = v = x$ or $u = v = y$, or $u = v = z$). Therefore, we can construct h-circles centered at x and y and u and v would be their two intersection points. By our earlier results we know that they are Euclidean circles and, since $u \neq v$, that u and v are mutually conjugate. The point z is an intersection of two h-circles centered at u and v, with the same radius. We know those two h-circles are Euclidean circles and by using the formula for the Euclidean center we can easily show that their Euclidean centers are also mutually conjugate. Then by simple reasoning about complex numbers (in Euclidean geometry) we can show that z must be real, and, therefore, h-collinear with $x = 0$ and y (that is on x-axis). H-collinearity then implies one of the three h-betweenness.

Negated Euclidean axiom.

```
lemma
"∃ a b c d t. {a, b, c, d, t} ⊆ unit_disc ∧
              h_betw a d t ∧ h_betw b d c ∧ a ≠ d ∧
              (∀ x y. x ∈ unit_disc ∧ y ∈ unit_disc ∧
                      h_betw a b x ∧ h_betw a c y ⟶ ¬ h_betw x t y)"
```

We show that the statement holds for points $a = 0$, $b = 1/2$, $c = i/2$, $d = (5 - \sqrt{17})/4 + i \cdot (5 - \sqrt{17})/4$, and $t = 1/2 + i/2$. H-betweenness of $a = 0$, d, and t is easily shown by showing that d and t have the same argument (since they are proportional by a positive real factor) and $|d| < |t|$. H-betweenness of b, d, and c is shown by showing that the h-distance between b and c is equal to the sum of h-distances between b and d and d and c. Since these two are the same, this reduces to showing that the h-distance between b and c is twice the h-distance between b and d. These are calculated by the h-distance explicit formula, derived earlier. By this formula, after brief calculations we determine that the h-distance between b and d is equal to $\operatorname{arccosh}(\sqrt{17}/3)$.

On the other hand, h-distance between b and c is equal to $\operatorname{arccosh}(25/9)$. The h-betweenness then follows by the formula $2\operatorname{arccosh} x = \operatorname{arccosh} 2x^2 - 1$, that holds for $x \geq 1$.

It remains to show that forall x and y, if t is h-between x and y, then it does not intersect either the positive segment of the x axis within the unit disc, or the positive segment of the y axis within the unit disc. We consider an arbitrary h-line through $t = 1/2 + i/2$. If it is given by parameters A and B, then by the circline equation it must hold that $3A + 2\operatorname{Re} B + 2\operatorname{Im} B = 0$. If it intersects the positive segment of the x-axis within the unit disc, it must hold that $A \neq 0$ and that $\operatorname{Re} B/\operatorname{Re} A < -1$. But then it holds that $\operatorname{Im} B/A = -\operatorname{Re} B/A - 3/2 > 1 - 3/2 = -1/2$. Therefore it cannot hold that $\operatorname{Im} B/A < -1$, what must hold if the h-line intersects the positive segment of the y-axis within the unit disc.

There are many equivalent statements for Euclid's parallel postulate. The equivalence between 34 postulates was previously formalized in Coq [11]. Choosing a different version of the postulate can help reducing the complexity and size of the proof that this version does not hold in our model. Thus, it would be interesting to obtain these equivalence proofs within Isabelle/HOL[6], since, after carefully analyzing the 34 postulates, the following one expressing that for any three non-collinear points a, b and c there exists a point equidistant from them, allowed us to obtain a proof which was much simpler and two times shorter:

```
lemma
"∃ a b c. a ∈ unit_disc ∧ b ∈ unit_disc ∧ c ∈ unit_disc ∧
          ¬ (h_collinear {a, b, c}) ∧
          ¬ (∃ x. x ∈ unit_disc ∧ h_cong a x b x ∧ h_cong a x c x)"
```

We showed that the statement holds for points $a = i/2$, $b = -i/2$ and $c = 1/5$. The first part was to prove that points a, b and c are not h-collinear. This was easily achieved since a and b determine unique line, the y-axis, and c does not lie on it.

The next part was to prove that perpendicular bisectors of segments ab and ac do not meet (and thus there does not exist a point equidistant from a, b and c). The perpendicular bisector of segment ab is determined by h-distance equation `h_dist a x = h_dist b x`. Points a and b are conveniently chosen, so that the perpendicular bisector of segment ab is the x-axis, thus greatly simplifying the calculation. Then, with some small calculation, we proved that no point on the x-axis satisfies the h-distance equation determining that a point belongs to perpendicular bisector of segment ac: `h_dist a x = h_dist c x`.

Existence of limiting parallels. We want to show that for each point a and each line x_1x_2 not containing a there exist two different points a_1 and a_2 so that rays aa_1 and aa_2 do not meet the line x_1x_2, and that for each point a' in angle a_1aa_2 the line aa' meets the line x_1x_2. To prove that this axiom holds in the

[6] There already exists tools that translate proofs from one proof assistant to another [1].

Poincare disc, we apply wlog principle twice. First, wlog we can assume that $a = 0$ (this can be achieved by a Blaschke transform). Next we denote by i_1 and i_2 the two different ideal points of the line x_1x_2. We show that $a_1 = i_1/2$ and $a_2 = i_2/2$ satisfy the axiom conclusion.

Since $i_1 \neq i_2$, it holds that $a_1 \neq a_2$.

In the extended complex plane, the ray $0a_1$ (it is an Euclidean ray) meets the line x_1x_2 (it is a part of an Euclidean circle) in the point i_1. The point i_1 is not within the unit disc. There cannot exist their another intersection point within the unit disc. If there would be such a point x, then both x_1x_2 and $0a_1$ would contain x, i_1 and the inverse of x (wrt. the unit circle), since both x_1x_2 and $0a_1$ are h-lines. These three points are different (since x is within, i_1 is on and the inverse of x is outside the unit disc), so x_1x_2 and $0a_1$ would be the same circline, that is impossible since $a = 0$ is not on the line x_1x_2. Therefore, x_1x_2 and $0a_1$ do not meet within the unit disc. The proof that x_1x_2 and $0a_2$ do not meet within the unit disc is analogous.



Fig. 9: Existence of limiting parallels – wlog position

Next we prove that for each point a' in the angle a_1aa_2 the ray $0a'$ meets the line x_1x_2. Since a' is in the angle a_1aa_2 there exists x such that it is h-between a_1 and a_2 and on the ray $0a'$. We again employ the wlog principle and assume that x is on the positive part of the x-axis (this can be achieved rotating the whole configuration around the origin). Therefore x is a positive real number. Line a_1a_2 is not equal to x-axis (otherwise, since, $a_1 = i_1/2$ and $a_2 = i_2/2$, i_1 and i_2 would have to be equal to -1 and 1, and 0 would lie on x_1x_2, which is false). Therefore it intersects the positive part of x-axis. The line x_1x_2 intersects x-axis. Namely, since x is between a_1 and a_2, and it is on the x-axis, it holds that $\operatorname{Im} a_1 \cdot \operatorname{Im} a_2 < 0$. Since, $\operatorname{Im} i_1 \cdot \operatorname{Im} i_2 = 4 \cdot \operatorname{Im} a_1 \cdot \operatorname{Im} a_2 < 0$, x_1x_2 does intersect the x-axis. We need to show that it intersects its positive part. Let $A_a = i \cdot (a_1\overline{a_2} - a_2\overline{a_1})$ and $B_a = i \cdot (a_2(|a_1|^2 + 1) - a_1(|a_2|^2 + 1))$ be the coefficients of the h-line a_1a_2, and let $A_i = i \cdot (i_1\overline{i_2} - i_2\overline{i_1})$ and $B_i = i \cdot (i_2(|i_1|^2 + 1) - i_1(|i_2|^2 + 1))$ be the coefficients of the h-line i_1i_2 i.e., the h-line x_1x_2. The coefficients A_a and A_i are real. Since a_1a_2 intersects the positive part of the x-axis, we know that $A_a \neq 0$ and $\operatorname{Re} B_a / \operatorname{Re} A_a < -1$. Since $a_1 = i_1/2$ and $a_2 = i_2/2$ by direct calculations it can be shown that

$$\frac{\operatorname{Re} B_i}{\operatorname{Re} A_i} = \frac{\operatorname{Im} i_1 - \operatorname{Im} i_2}{\operatorname{Im} i_2 \cdot \operatorname{Re} i_1 - \operatorname{Im} i_1 \cdot \operatorname{Re} i_2} = \frac{4}{5} \frac{\operatorname{Re} B_a}{\operatorname{Re} A_a}$$

so, since $\operatorname{Re} B_a / \operatorname{Re} A_a$ is negative, so is $\operatorname{Re} B_i / \operatorname{Re} A_i$.

Since 0 is not on the line $x_1 x_2$, it is not an Euclidean line and $A_i \neq 0$. Since $x_1 x_2$ intersects the x-axis, it must hold that $(\operatorname{Re} B_i / \operatorname{Re} A_i)^2 > 1$. Therefore $\operatorname{Re} B_i / \operatorname{Re} A_i < -1$, and the line $x_1 x_2$ intersects the positive part of x-axis (i.e., the ray $0x$ which is the ray aa').

Continuity axiom. Tarski's continuity axiom is essentially the Dedekind cut construction. Intuitively, if all points of a set of points are on one side of all points of the other set of points, then there is a point between the two sets. The original Tarski's axioms are defined within the framework of First Order Logic and sets are not explicitly recognized in Tarski's formalization. Instead of speaking about sets of points, Tarski uses first order predicates ϕ and ψ. However, while it would be possible to express the restriction of predicates f and g to FOL predicates within the Higher Order Logic framework of Isabelle/HOL as in [31], we chose to avoid it. Therefore, from a strict viewpoint, our formalization of Tarski's axioms within Isabelle/HOL gives a different geometry then Tarski's original axiomatic system.

```
lemma continuity_axiom:
assumes "∃ a ∈ unit_disc. ∀ x ∈ unit_disc. ∀ y ∈ unit_disc.
          φ x ∧ ψ y ⟶ h_betw a x y"
shows "∃ b ∈ unit_disc. ∀ x ∈ unit_disc. ∀ y ∈ unit_disc.
          φ x ∧ ψ y ⟶ h_betw x b y"
```

Still, it turns out that it is possible to show that Poincaré model also satisfies the stronger variant of the axiom (without FOL restrictions on predicates ϕ and ψ). If one of the sets is empty, the statement trivially holds. If the sets have a point in common, that point is the point sought. In other cases, the proof is obtained using wlog reasoning. Möbius transformations that preserve the unit disc are applied so that all points, that satisfies both predicates, lie on the positive part of x-axis (and inside unit disc). Proving this requires using non-trivial properties of reals, i.e., their completeness. Completeness of reals in Isabelle/HOL is formalized in the following theorem (the supremum, i.e., the least upper bound property)[7]:

```
lemma
assumes "x ∈ X" "∃ M. ∀ x ∈ X. x ≤ M"
shows "x ≤ Sup X"
```

The supremum property of reals can be used since all points are on the x-axis and hence their imaginary part is equal zero. Hence, all points can

[7] there is no development about real closed fields in Isabelle/HOL. This is the reason that motivated us to avoid the restriction of predicates f and g to FOL predicates.

be observed as reals. Then, the statement reduces to proving that the first set of points (satisfying the first predicate) has a supremum and that this supremum is the point sought. The first and the second sets of points do not have a common point (one of the special cases in the beginning of proof), so the second set of points serves as upper bound for the first set of points.

Putting everything together, we could prove that the Poincaré disc model form a model of elementary geometry:

interpretation TarskiAbsolute $\equiv_h$ $\mathfrak{B}_h$
interpretation TarskiHyperbolic $\equiv_h$ $\mathfrak{B}_h$
interpretation ElementaryTarskiHyperbolic $\equiv_h$ $\mathfrak{B}_h$

5 Conclusions and further work

We have formally defined the Poincaré disc model within the projective complex line $\mathbb{C}P^1$, and mechanically checked that it satisfies Tarski's axioms except for Euclid's axiom. Contrary to the popular belief that it is much harder to do than for the Klein-Beltrami model [40], our experience shows that there are both advantages and disadvantages for both models. Definitely, the hardest thing in the Poincaré model was to show that it satisfies Pasch's axiom, since it requires introducing the betweenness on circlines (while the Klein model inherits the betweenness on the Euclidean line segments). However, most other axioms are verified more easily than for the Klein model, due to the fact that circles in the Poincaré model are Euclidean circles, leading to a very easy proof of existence of a unique h-isometry between two congruent triangles. The negated Euclidean axiom was easier to prove in one of its equivalent formulations, showing the usefulness of having several equivalent formulations of axioms [11].

Many proofs have been done by algebraic calculations. Simplest lemmas were proved automatically, by employing reasoning about fields (available in the `field_simps` collection of lemmas). In several lemmas the tatic `algebra` (based on Gröbner basis method) was very useful and helped in reducing size of some proofs.

The central technique, without which the formalization would not have been possible, was *without loss of generality reasoning* [33]. We have proved 14 different wlog lemmas and used them abundantly. Basing the model in the extended complex plane (instead of the real plane) and using the well-developed machinery of complex vectors and matrices (wonderfully described by Schwerdfeger [61] and formalized in the previous work of the first two authors [41]) was also a very important decision, leading to nicer and more elegant formalization (for example, we never needed to distinguish between the line and circle segments within the Poincaré disc, that is ubiquitous in expositions based on classic, Euclidean and Cartesian geometry).

The formalization of the Poincaré disc model can go beyond showing that it satisfies the axioms. For example, based on our foundation, hyperbolic trigonometry can be developed (we have already easily proved the Pythagorean theorem and the hyperbolic law of cosines), and Gauss-Bolyai-Lobachevsky formula (connecting the angle of parallelism with the distance) can be proved.

We plan to extend our work to other models of hyperbolic geometry. Although proving that one model satisfies the axioms simplifies that proof for the other model (as one can apply transformations from one model to the other one), defining other models and their basic objects requires some non-trivial effort.

Although the exposition in the present paper was driven by our Isabelle/HOL formalization, we hope that it can be useful to a wider audience interested in formal geometry.

References

1. Ali Assaf. *A framework for defining computational higher-order logics*. Theses, École polytechnique, September 2015.
2. Clemens Ballarin. Interpretation of locales in Isabelle: Theories and proof contexts. In *Mathematical Knowledge Management, MKM, Proceedings*, pages 31–43, 2006.
3. Michael Beeson. Proving Hilbert's axioms in Tarski geometry. `http://www.michaelbeeson.com/research/papers/TarskiProvesHilbert.pdf`, 2014.
4. Michael Beeson. A Constructive Version of Tarski's Geometry. *Annals of Pure and Applied Logic*, 166(11):1199–1273, 2015.
5. Michael Beeson, Pierre Boutry, Gabriel Braun, Charly Gries, and Julien Narboux. GeoCoq, June 2018.
6. Michael Beeson and Larry Wos. Finding Proofs in Tarskian Geometry. *Journal of Automated Reasoning*, 58(1):181–207, 2017.
7. Eugenio Beltrami. *Saggio di interpretazione della geometria Non-Euclidea*. s.n., 1868.
8. Janos Bolyai. *Appendix, Scientiam Spatii absolute veram exhibens: a veritate aut falsitate Axiomatis XI. Euclidei (a priori haud unquam decidenda) independentem; adjecta ad casum falsitatis, quadratura circuli geometrica. Auctore Johanne Bolyai de eadem, Geometrarum in Exercitu Caesareo Regio Austriaco Castrensium Capitaneo*. Coll. Ref., 1832.
9. Karol Borsuk and Wanda Szmielew. *Foundations of Geometry*. North-Holland, 1960.
10. Pierre Boutry, Gabriel Braun, and Julien Narboux. Formalization of the Arithmetization of Euclidean Plane Geometry and Applications. *Journal of Symbolic Computation*, 2018.
11. Pierre Boutry, Charly Gries, Julien Narboux, and Pascal Schreck. Parallel Postulates and Continuity Axioms: A Mechanized Study in Intuitionistic Logic Using Coq. *Journal of Automated Reasoning*, Sep 2017.
12. Pierre Boutry, Julien Narboux, Pascal Schreck, and Gabriel Braun. A short note about case distinctions in Tarski's geometry. In Francisco Botana and Pedro Quaresma, editors, *Proceedings of the Tenth International Workshop on Automated Deduction in Geometry*, Proceedings of ADG 2014, pages 51–65, Coimbra, Portugal, July 2014.
13. David Braun, Nicolas Magaud, and Pascal Schreck. An Equivalence Proof Between Rank Theory and Incidence Projective Geometry. In *Proceedings of ADG 2016*, pages 62–77, 2016.
14. Gabriel Braun, Pierre Boutry, and Julien Narboux. From Hilbert to Tarski. In Julien Narboux, Pascal Schreck, and Ileana Streinu, editors, *Proceedings of the Eleventh International Workshop on Automated Deduction in Geometry*, Proceedings of ADG 2016, pages 78–96, Strasbourg, France, June 2016.
15. Gabriel Braun and Julien Narboux. From Tarski to Hilbert. In Tetsuo Ida and Jacques Fleuriot, editors, *Automated Deduction in Geometry (ADG 2012)*, volume 7993 of *Lecture Notes in Computer Science*, pages 89–109. Springer, Edinburgh, United Kingdom, September 2012.
16. Gabriel Braun and Julien Narboux. A synthetic proof of Pappus' theorem in Tarski's geometry. *Journal of Automated Reasoning*, 58(2):23, February 2017.

17. Christophe Brun, Jean-François Dufourd, and Nicolas Magaud. Formal Proof in Coq and Derivation of a Program in C++ to Compute Convex Hulls. In Tetsuo Ida and Jacques Fleuriot, editors, *Automated Deduction in Geometry (ADG 2012)*, volume 7993 of *Lecture Notes in Computer Science*, pages 71–88. Springer Verlag, Sep 2012.
18. Roland Coghetto. Klein-Beltrami Model. Part I. *Formalized Mathematics*, 26(1):21–32, 2018.
19. Roland Coghetto. Klein-Beltrami Model. Part II. *Formalized Mathematics*, 26(1):33–48, 2018.
20. Coxeter, Harold Scott Macdonald. *Projective Geometry.* Springer Science & Business Media, 2003.
21. Vincenzo De Risi. The development of Euclidean axiomatics. *Archive for History of Exact Sciences*, 70(6):591–676, Nov 2016.
22. Christophe Dehlinger and Jean-François Dufourd. Formalizing generalized maps in Coq. *Theoretical Computer Science*, 323(1):351 – 397, 2004.
23. Christophe Dehlinger, Jean-François Dufourd, and Pascal Schreck. Higher-Order Intuitionistic Formalization and Proofs in Hilbert's Elementary Geometry. In *Automated Deduction in Geometry (ADG 2000)*, volume 2061 of *Lecture Notes in Computer Science*, pages 306–324, 2001.
24. Jean-François Dufourd. A Hypermap Framework for Computer-aided Proofs in Surface Subdivisions: Genus Theorem and Euler's Formula. In *Proceedings of the 2007 ACM Symposium on Applied Computing*, SAC '07, pages 757–761, New York, NY, USA, 2007. ACM.
25. Jean-François Dufourd and Yves Bertot. Formal study of plane Delaunay triangulation. In Lawrence Charles Matt Kaufmann Paulson, editor, *Interactive Theorem Proving'2010 (In Federative Logic Conference: FLoC'2010)*, number 6172 in Lecture Notes in Computer Science, pages 211–226. Springer-Verlag, Jul 2010.
26. Jacques Fleuriot. Nonstandard Geometric Proofs. In Jürgen Richter-Gebert and Dongming Wang, editors, *Automated Deduction in Geometry (ADG 2000)*, pages 246–267, Berlin, Heidelberg, 2001. Springer Berlin Heidelberg.
27. Jacques Fleuriot. Theorem Proving in Infinitesimal Geometry. *Logic Journal of the IGPL*, 9(3):447–474, 2001.
28. Jacques Fleuriot. Exploring the Foundations of Discrete Analytical Geometry in Isabelle/HOL. In Pascal Schreck, Julien Narboux, and Jürgen Richter-Gebert, editors, *Automated Deduction in Geometry (ADG 2010)*, volume 6877 of *Lecture Notes in Computer Science*, pages 34–50. Springer, 2010.
29. Hans Freudenthal et al. K. borsuk and wanda szmielew, foundations of geometry, euclidean and bolyai-lobachevskian geometry, projective geometry. *Bulletin of the American Mathematical Society*, 67(4):342–344, 1961.
30. Carl Friedrich Gauss. *Disquisitiones generales circa superficies curvas.* [Mathematical tracts. Typis Dieterichianis, 1828.
31. Charly Gries, Julien Narboux, and Pierre Boutry. Axiomes de continuité en géométrie neutre : une étude mécanisée en Coq. In Nicolas Magaud and Zaynah Dargaye, editors, *Journées Francophones des Langages Applicatifs 2019*, Acte des Journées Francophones des Langages Applicatifs (JFLA 2019), Les Rousses, France, January 2019.
32. Haragauri Narayan Gupta. *Contributions to the Axiomatic Foundations of Geometry.* PhD thesis, University of California, Berkley, 1965.
33. John Harrison. Without loss of generality. In Stefan Berghofer, Tobias Nipkow, Christian Urban, and Makarius Wenzel, editors, *Theorem Proving in Higher Order Logics*, pages 43–59, Berlin, Heidelberg, 2009. Springer Berlin Heidelberg.
34. Brian Huffman and Ondřej Kunčar. Lifting and Transfer: A Modular Design for Quotients in Isabelle/HOL. In Georges Gonthier and Michael Norrish, editors, *Certified Programs and Proofs*, pages 131–146. Springer International Publishing, Cham, 2013.
35. Gilles Kahn. Constructive geometry according to Jan von Plato, 1995.
36. Nicolaus Lobatschewsky. *Geometrische Untersuchungen zur Theorie der Parallellinien*, pages 159–223. Springer Vienna, Vienna, 1985.
37. Johann Friedrich Lorenz. *Grundriss der reinen und angewandten Mathematik.* Fleckeisen, 1791.

38. Nicolas Magaud, Agathe Chollet, and Laurent Fuchs. Formalizing a Discrete Model of the Continuum in Coq from a Discrete Geometry Perspective. *Annals of Mathematics and Artificial Intelligence*, pages 309–332, January 2015.
39. Nicolas Magaud, Julien Narboux, and Pascal Schreck. A Case Study in Formalizing Projective Geometry in Coq: Desargues Theorem. *Computational Geometry*, 45(8):406–424, 2012.
40. Timothy James McKenzie Makarios. A Mechanical Verification of the Independence of Tarski's Euclidean Axiom. Master's thesis, Victoria University of Wellington, 2012. Master Thesis.
41. Filip Marić and Danijela Simić. Formalizing Complex Plane Geometry. *Annals of Mathematics and Artificial Intelligence*, 74(3-4):271–308, 2015.
42. Filip Marić and Danijela Simić. Complex geometry. *Archive of Formal Proofs*, December 2019. `http://isa-afp.org/entries/Complex_Geometry.html`, Formal proof development.
43. Andrew McFarland, Joanna McFarland, and James Smith, editors. *Alfred Tarski: Early Work in Poland - Geometry and Teaching.* Springer New York, 2014.
44. Laura Meikle and Jacques. Fleuriot. Formalizing Hilbert's Grundlagen in Isabelle/Isar. In *Theorem Proving in Higher Order Logics*, pages 319–334, 2003.
45. Laura Meikle and Jacques Fleuriot. Mechanical Theorem Proving in Computational Geometry. In Hoon Hong and Dongming Wang, editors, *Automated Deduction in Geometry (ADG 2004)*, pages 1–18, Berlin, Heidelberg, 2006. Springer Berlin Heidelberg.
46. Julien Narboux. Mechanical Theorem Proving in Tarski's geometry. In Francisco Botana Eugenio Roanes Lozano, editor, *Automated Deduction in Geometry (ADG 2006)*, volume 4869 of *Lecture Notes in Computer Science*, pages 139–156. Springer, Pontevedra, Spain, 2007.
47. Julien Narboux, Predrag Janičić, and Jacques Fleuriot. Computer-assisted Theorem Proving in Synthetic Geometry. In Meera Sitharam, Audrey St. John, and Jessica Sidman, editors, *Handbook of Geometric Constraint Systems Principles*, chapter 2, pages 25–73. Chapman and Hall/CRC, 2018. in press.
48. Tristan Needham. *Visual Complex Analysis.* Oxford University Press, 1998.
49. Tobias Nipkow. Programming and Proving in Isabelle/HOL, 2013.
50. Tobias Nipkow, Lawrence C Paulson, and Markus Wenzel. *Isabelle/HOL: a Proof Assistant for Higher-Order Logic*, volume 2283. Springer Science & Business Media, 2002.
51. Moritz Pasch. *Vorlesung über Neuere Geometrie.* Springer, 1976.
52. Lawrence C. Paulson. Natural Deduction as Higher-Order Resolution. *The Journal of Logic Programming*, 3(3):237 – 258, 1986.
53. Lawrence C. Paulson. The Foundation of a Generic Theorem Prover. *Journal of Automated Reasoning*, 5(3):363–397, Sep 1989.
54. David Pichardie and Yves Bertot. Formalizing Convex Hull Algorithms. In Richard Boulton and Paul Jackson, editors, *Theorem Proving in Higher Order Logics*, pages 346–361, Berlin, Heidelberg, 2001. Springer Berlin Heidelberg.
55. François Puitg and Jean-François Dufourd. Formal specification and theorem proving breakthroughs in geometric modeling. In Jim Grundy and Malcolm Newey, editors, *Theorem Proving in Higher Order Logics*, pages 401–422, Berlin, Heidelberg, 1998. Springer Berlin Heidelberg.
56. William Richter. Formalizing Rigorous Hilbert Axiomatic Geometry Proofs in the Proof Assistant Hol Light. `http://citeseerx.ist.psu.edu/viewdoc/download?doi=10.1.1.363.2373&rep=rep1&type=pdf`.
57. William Richter, Adam Grabowski, and Jesse Alama. Tarski Geometry Axioms. *Formalized Mathematics*, 22(**2**):167–176, 2014.
58. Bernhard Riemann and Hermann Weyl. *Über die Hypothesen, welche der Geometrie zu Grunde liegen.* Springer, 1921.
59. Giovanni Girolamo Saccheri. *Euclides ab omni naevo vindicatus.* Mediolani : Ex typographia Pauli Antonio Montani, 1733.
60. Wolfram Schwabhäuser, Wanda Szmielew, and Alfred Tarski. *Metamathematische Methoden in der Geometrie.* Springer-Verlag, Berlin, 1983.
61. Hans Schwerdtfeger. *Geometry of Complex Numbers: Circle Geometry, Moebius Transformation, Non-euclidean Geometry.* Courier Corporation, 1979.

62. Phil Scott. Mechanising Hilbert's foundations of geometry in Isabelle. Master's thesis, University of Edinburgh, 2008. Master Thesis.
63. Phil Scott and Jacques Fleuriot. An Investigation of Hilbert's Implicit Reasoning through Proof Discovery in Idle-Time. In *Proceedings of the Seventh International Workshop on Automated Deduction in Geometry*, pages 182–200, 2010.
64. Danijela Simić, Filip Marić, and Pierre Boutry. Poincaré disc model. *Archive of Formal Proofs*, December 2019. `http://isa-afp.org/entries/Poincare_Disc.html`, Formal proof development.
65. Sana Stojanović Đurđević, Julien Narboux, and Predrag Janičić. Automated Generation of Machine Verifiable and Readable Proofs: A Case Study of Tarski's Geometry. *Annals of Mathematics and Artificial Intelligence*, pages 249–269, 2015.
66. Alfred Tarski and Steven Givant. Tarski's System of Geometry. *The Bulletin of Symbolic Logic*, 5(2):175–214, June 1999.
67. Jan von Plato. The Axioms of Constructive Geometry. *Annals of Pure and Applied Logic*, 76(2):169 – 200, 1995.
68. Markus Wenzel. *Isabelle/Isar — a versatile environment for human-readable formal proof documents*. PhD thesis, Technische Universität München, 2002.